

12 Boolean Algebra

Introduction

In this chapter, we will use Maple to model Boolean algebra. In the first section, we demonstrate the basic commands of the **Logic** package, which will be used extensively in this chapter. In the second section, we will focus on the disjunctive normal form of a logical expression. We will see how to use Maple's command for finding a disjunctive normal form expression for a Boolean function, and we will develop a procedure for finding such a representation for a function defined by a table of values. In Section 3, we will see how Maple can be used to model logical circuits, including how to go about transforming a circuit diagram into a Maple expression. We also provide a procedure that will transform a logical expression into a model of a circuit. In the final section of the chapter, we consider simplification of logical expressions, and we develop an implementation of the Quine–McCluskey method.

In this chapter, we will be using the Maple package **Logic**. This package includes several commands related to Boolean algebra that will be useful. We load this package now.

```
> with(Logic):
```

12.1 Boolean Functions

In this section, we will introduce Maple's **Logic** package, which can be used to explore Boolean algebra. In particular, we will see how Maple represents Boolean operators, how to work with Boolean expressions, and how to create Boolean functions. We will also use Maple to verify identities in Boolean algebra and to compute the dual of an expression.

Preliminaries

In Chapter 1 of this manual, we discussed Maple's logical expressions. The Boolean values *true* and *false* are represented by the literal constants **true** and **false**. To Maple, these are constant values, like the numbers 2 or **Pi**.

We also saw in Chapter 1 the logical operators, including **and**, **or**, and **not**. These are similar to arithmetic operators like + and *. Combining Boolean values with the logical operators causes Maple to evaluate the resulting expression.

```
> true or (not(false) and false)
true
```

(12.1)

These “ordinary” operators are a vital part of any programming language, as they are needed for controlling execution of procedures. Moreover, Maple recognizes a third value, **FAIL**, which is useful from the perspective of programming. (The help page for Boolean expressions provides tables showing how the operations are defined on **true**, **false**, and **FAIL**.)

The **Logic** package provides a second set of Boolean operators. The basic operators are **&and**, **&or**, and **¬**. These are supplemented by **&nand**, **&nor**, **&xor** (exclusive or), **&implies** (implication), and **&iff** (biconditional).

These operators are different from the ordinary operators in three ways. First, they are inert, as opposed to active. This means that when you enter an expression in Maple using the **Logic** package operators, they are not immediately evaluated. They are, however, displayed in symbol form in the output.

$$\begin{aligned} &> \text{ (true \&or \¬(false))\&and false} \\ &\quad (true \vee (\neg false)) \wedge false \end{aligned} \tag{12.2}$$

This makes it easier to explore and analyze Boolean expressions symbolically since Maple will not perform simplification until you explicitly tell it to do so.

The second difference is that the operators in the **Logic** package do not recognize **FAIL** as a logical value.

The third difference is that the **Logic** operators all have equal precedence, so you should fully parenthesize statements.

More on Precedence

Precedence and parentheses require a bit of explanation. Recall that the usual order of operations for logical operators is **not**, then **and**, then **or**, and finally implication. When you enter an expression with the **Logic** operators, this order of precedence is not respected.

For example, if you enter p **or** q **and** r , using the **Logic** operators, Maple will consider the operations from left to right.

$$\begin{aligned} &> p \text{ \&or } q \text{ \&and } r \\ &\quad (p \vee q) \wedge r \end{aligned} \tag{12.3}$$

Note that in the output for that expression, Maple has put $p \vee q$ in parentheses. This is consistent with the fact that the operators have equal precedence, meaning that the **&or** is applied first. That is, $(p \vee q) \wedge r$ is the interpretation of what was input.

Using the inert **Logic** operators, you must enforce the order of precedence yourself. To input $p \vee q \wedge r$ and have Maple interpret it in the correct order, you must use parentheses.

$$\begin{aligned} &> p \text{ \&or } (q \text{ \&and } r) \\ &\quad p \vee (q \wedge r) \end{aligned} \tag{12.4}$$

It is a good idea to always fully parenthesize your input with the **Logic** package or you may obtain erroneous results.

The 0–1 Form of Boolean Algebra

We conclude this subsection with a warning.

The textbook uses the objects 0 and 1 with operations $+$, \cdot , and $-$ instead of *true*, *false*, \vee , \wedge , and \neg as we have done. Maple does, in fact, have commands available for using the 0–1 form. The **Logic** package contains commands, **Import** and **Export**, that can be used to change between expressions involving the **Logic** inert operators and either the active Boolean operators or a 0–1 form, called **MOD2**.

However, the **MOD2** form is in opposition to the conventions used in the textbook. Specifically, in Maple, **+** corresponds to **&xor**, rather than **&or**.

To avoid the confusion that this difference could create, we will always use the logical form in this manual. However, some readers may be interested to know that it is possible to create operators to mirror the kinds of Boolean algebra expressions used in the text. To do so, you can create definitions for the inert addition, multiplication, and negation operators. Recall that in Chapter 4, we made definitions for the inert arithmetic operators in order to emulate modular arithmetic.

For example, to define a Boolean **+**, we define **&+** as follows.

$$\begin{aligned} > \text{'\&+'} := (a :: \{0, 1\}, b :: \{0, 1\}) \rightarrow \text{ifelse}(a = 1 \text{ or } b = 1, 1, 0) \\ &\quad \text{'\&+'} := (a :: \{0, 1\}, b :: \{0, 1\}) \mapsto \text{ifelse}(a = 1 \text{ or } b = 1, 1, 0) \end{aligned} \quad (12.5)$$

The type declarations will ensure that only bits are allowed to be addends.

With this definition, we can now compute $0 + 1$.

$$\begin{aligned} > 0 \&+ 1 \\ &\quad 1 \end{aligned} \quad (12.6)$$

Once the definitions of **&*** and a unary **&-** are in place, we could compute the value of $1 \cdot 0 + (0 + 1)$ (Example 1) by entering the following:

$$> (1 \&* 0) \&+ (\&-(0 \&+ 1))$$

Definitions of the other operations are left to the interested reader. In this manual, we will not use this approach, since the logical form of Boolean expressions is more naturally supported by Maple.

Boolean Expressions and Boolean Functions

We saw above how to create Boolean expressions using the inert **Logic** operators. Now, we will look at how to evaluate Boolean expressions and how to create Boolean functions.

Evaluating Boolean Expressions

Because the operators in **Logic** are inert, you must explicitly tell Maple to evaluate expressions involving them by applying the **BooleanSimplify** command.

Consider Example 1 from the text, which asks that we compute the value of $1 \cdot 0 + \overline{(0 + 1)}$. To perform this computation in Maple, we must first translate it into a logical statement. We do this by changing 1 into *true*, 0 into *false*, the multiplication into \wedge , the addition into \vee , and the bar into \neg .

This results in the Boolean expression $(\text{true} \wedge \text{false}) \vee \neg(\text{false} \vee \text{true})$. Note that we added parentheses since the **Logic** arguments have equal precedence.

Using Maple's active logical operators, as in Chapter 1, we would just enter the statement to evaluate it.

$$\begin{aligned} > (\text{true and false}) \text{ or not}(\text{false or true}) \\ &\quad \text{false} \end{aligned} \quad (12.7)$$

Using the inert operators, we apply the **BooleanSimplify** command. This command accepts only one argument, the logical expression that it is to simplify. In this example, it will simplify the expression down to a single truth value.

> **BooleanSimplify**((*true* &and *false*) &or ¬(*false* &or *true*))
false (12.8)

As you might guess from the name, **BooleanSimplify** does more than evaluate an expression to a truth value. It will also simplify more general expressions involving unassigned names. For example, consider the Boolean expression $(x + y)(x + z)$. In logical terms, this is $(x \vee y) \wedge (x \vee z)$.

> **BooleanSimplify**((*x* &or *y*) &and (*x* &or *z*))
 $x \vee (y \wedge z)$ (12.9)

Note that Maple has simplified this in accordance with the distributive law.

Representing Boolean Functions

Defining a Boolean function is identical to defining any other function in Maple.

Consider, for example, the Boolean function shown below (written in the 0–1 notation).

$$F(x, y, z) = xy + yz + zx.$$

This can be modeled in Maple by the procedure **Fp** (p for procedure).

```
1 Fp := proc (x, y, z)
2   return (x &and y) &or (y &and z) &or (z &and x) ;
3 end proc;
```

It is often more natural, however, to model such functions as functional operators.

> **F** := (*x, y, z*) → (*x* &and *y*) &or (*y* &and *z*) &or (*z* &and *x*)
 $F := (x, y, z) \mapsto ((x \wedge y) \vee (y \wedge z)) \vee (z \wedge x)$ (12.10)

You can work with **F** (or **Fp**) in the usual way. The following applies **F** to **p**, **q**, and **r**.

> **F**(*p, q, r*)
 $((p \wedge q) \vee (q \wedge r)) \vee (r \wedge p)$ (12.11)

Observe what happens when **F** is applied to truth values.

> **F**(*true, false, true*)
 $((true \wedge false) \vee (false \wedge true)) \vee (true \wedge true)$ (12.12)

Because the operators **&and** and **&or** are inert, Maple does not automatically simplify to a truth value. We have to explicitly call **BooleanSimplify**.

> **BooleanSimplify**(**F**(*true, false, true*))
true (12.13)

It can be easier to embed the call to **BooleanSimplify** in the function definition itself. We redefine **F** to include **BooleanSimplify**.

$$\begin{aligned}
 &> F := (x, y, z) \rightarrow \\
 &\quad \text{BooleanSimplify}((x \text{ \&and } y) \text{ \&or } (y \text{ \&and } z) \text{ \&or } (z \text{ \&and } x)) \\
 &F := (x, y, z) \mapsto \text{BooleanSimplify}(((x \wedge y) \vee (y \wedge z)) \vee (z \wedge x))
 \end{aligned} \tag{12.14}$$

Now applying **F** to **p**, **q**, and **r** produces the same result as before, since the expression cannot be simplified any further.

$$\begin{aligned}
 &> F(p, q, r) \\
 &\quad (p \wedge q) \vee (q \wedge r) \vee (r \wedge p)
 \end{aligned} \tag{12.15}$$

But applying **F** to truth values will return a single truth value.

$$\begin{aligned}
 &> F(\text{true}, \text{false}, \text{true}) \\
 &\quad \text{true}
 \end{aligned} \tag{12.16}$$

You can also mix truth values and symbols. In this case, the inclusion of **BooleanSimplify** in the definition of **F** causes Maple to simplify the expression as much as possible, given the partial information.

$$\begin{aligned}
 &> F(\text{true}, q, r) \\
 &\quad q \vee r
 \end{aligned} \tag{12.17}$$

The output indicates that if p is known to be true, then $(p \wedge q) \vee (q \wedge r) \vee (r \wedge p)$ is equivalent to $q \vee r$.

Values of Boolean Functions

Examples 4 and 5 of Section 12.1 illustrate how the values of a Boolean function, in the 0–1 format, can be displayed in a table. In the logical form, this is equivalent to a truth table for the Boolean function. In Chapter 1 of this manual, we created truth tables by looping through all the possible values and also mentioned **TruthTable**.

The **TruthTable** command requires two arguments. The first argument is a Boolean expression. The second is a list of the variable names that appear in the expression.

For example, we will display the table of values for the Boolean function **F** defined above. The first argument to the **TruthTable** command will be the Boolean expression obtained by **F(p,q,r)**. The second argument will be the list **[p,q,r]**.

$$> Fdataframe := \text{TruthTable}(F(p, q, r), [p, q, r])$$

$$\begin{bmatrix}
 & p & q & r & value \\
 1 & false & false & false & false \\
 2 & false & false & true & false \\
 3 & false & true & false & false \\
 4 & false & true & true & true \\
 5 & true & false & false & false \\
 6 & true & false & true & true \\
 7 & true & true & false & true \\
 8 & true & true & true & true
 \end{bmatrix}
 \quad (12.18)$$

Note that the result of **TruthTable** is a **DataFrame** object. If you wish to access individual elements, you can do so by referring to the row and column labels. For example, the entries in the 4th row corresponding to *q* and to the result are:

$$\begin{aligned}
 &> Fdataframe[4, 'q'] \\
 &\quad true
 \end{aligned}
 \quad (12.19)$$

$$\begin{aligned}
 &> Fdataframe[4, 'value'] \\
 &\quad true
 \end{aligned}
 \quad (12.20)$$

Note that the single quotes are not necessary, unless the name has been assigned to something.

Perhaps surprisingly, if you use the bracket selection operator with a single entry, either by number or name, it will refer to a column, not a row.

$$\begin{aligned}
 &> Fdataframe[4], Fdataframe['value'] \\
 &\quad \begin{bmatrix} 1 & false \\ 2 & false \\ 3 & false \\ 4 & true \\ 5 & false \\ 6 & true \\ 7 & true \\ 8 & true \end{bmatrix}, \begin{bmatrix} 1 & false \\ 2 & false \\ 3 & false \\ 4 & true \\ 5 & false \\ 6 & true \\ 7 & true \\ 8 & true \end{bmatrix}
 \end{aligned}
 \quad (12.21)$$

To extract a row, give two dots as the second index in the selection operation, which Maple understands to be the entire range of values. Below, we also put the row number in a list. This can be used to select multiple rows; here, it has the effect of the output appearing as a row.

$$\begin{aligned}
 &> Fdataframe[[4], ..] \\
 &\quad \begin{bmatrix}
 & p & q & r & value \\
 4 & false & true & true & true
 \end{bmatrix}
 \end{aligned}
 \quad (12.22)$$

The **TruthTable** command allows you to specify the representation of the truth table with the output option, if you would prefer a **Matrix** or **table**. Tables can be a very useful alternative to a **DataFrame**, since the indices in the table will be tuples corresponding to the values of the independent variables and the corresponding values are the truth values of the expression.

```
> Ftable := TruthTable(F(p,q,r),[p,q,r],output = table)
      Ftable := table([(true,false,false) = false, (false,false,true) = false,
      (true,false,true) = true, (true,true,false) = true,
      (true,true,true) = true, (false,true,true) = true,
      (false,true,false) = false, (false,false,false) = false])
```

(12.23)

We illustrate interacting with the table form by printing one entry at a time. Recall that **indices** applied to a **table** produces a sequence of lists.

```
> indices(Ftable)
      [true,false,false], [false,false,true], [true,false,true], [true,true,false],
      [true,true,true], [false,true,true], [false,true,false],
      [false,false,false]
```

(12.24)

In order to use the indices with the selection operation, we must apply **op** to remove the added list structure.

```
> for i in indices(Ftable) do
      print(i, Ftable[op(i)])
end do
      [true,false,false], false
      [false,false,true], false
      [true,false,true], true
      [true,true,false], true
      [true,true,true], true
      [false,true,true], true
      [false,true,false], false
      [false,false,false], false
```

(12.25)

Operations on Boolean Functions

As with functions on real numbers, Boolean functions can be combined using basic operations. The complement as well as the sum and product of Boolean functions are defined in the text.

To compute complements, sums, and products of Boolean functions, you must define a new function in terms of the original. For example, consider the function $G(x,y) = x \cdot y$. In logical notation, this is $G(x,y) = x \wedge y$.

```
> G := (x,y) → BooleanSimplify(x &and y)
      G := (x,y) ↦ BooleanSimplify(x ∧ y)
```

(12.26)

The complement of **G**, which we will call **notG**, is created as follows. The arguments of **notG** are the same as **G**. The formula that defines **notG** is **¬ G(x,y)**. That is, **¬** is applied to the result of evaluating **G** on the arguments. And once again, **BooleanSimplify** is called.

$$\begin{aligned}
> \text{not}G &:= (x, y) \rightarrow \text{BooleanSimplify}(\&\text{not}(G(x, y))) \\
\text{not}G &:= (x, y) \mapsto \text{BooleanSimplify}(\neg G(x, y))
\end{aligned} \tag{12.27}$$

Observe that if we evaluate **notG** at a pair of variables, the presence of **BooleanSimplify** ensures that De Morgan's law is applied.

$$\begin{aligned}
> \text{not}G(x, y) \\
&(\neg x) \vee (\neg y)
\end{aligned} \tag{12.28}$$

Let us define another function, $H(x, y) = x \cdot \bar{y}$.

$$\begin{aligned}
> H &:= (x, y) \rightarrow \text{BooleanSimplify}(x \&\text{and}(\&\text{not} y)) \\
H &:= (x, y) \mapsto \text{BooleanSimplify}(x \wedge (\neg y))
\end{aligned} \tag{12.29}$$

To compute the Boolean sum $G + H$, we combine the functions with the **&or** operator. More precisely, we define a functional operator **GpH** with the formula **G(x,y) &or H(x,y)**.

$$\begin{aligned}
> \text{GpH} &:= (x, y) \rightarrow \text{BooleanSimplify}(G(x, y) \&\text{or} H(x, y)) \\
\text{GpH} &:= (x, y) \mapsto \text{BooleanSimplify}(G(x, y) \vee H(x, y))
\end{aligned} \tag{12.30}$$

Applying this to a pair of variables, we obtain the following formula for $G + H$.

$$\begin{aligned}
> \text{GpH}(x, y) \\
&x
\end{aligned} \tag{12.31}$$

This result indicates that $x \cdot y + x \cdot \bar{y} = x$. This can also be verified using the identities in Table 5 of Section 12.1.

Identities of Boolean Algebra

The **Logic** package's **Equivalent** command makes checking identities and equivalence of Boolean expressions fairly straightforward. This command can also be used to check equality of Boolean functions.

Identities and Equivalence of Boolean Expressions

We will use the distributive law $x(y + z) = xy + xz$ as an example. First, we must translate the statement from the 0–1 form into a statement of logic: $x \wedge (y \vee z) \equiv (x \wedge y) \vee (x \wedge z)$.

Now, we assign the expressions on either side of the equivalence to names. This is not necessary, but it will make later statements easier to read.

$$\begin{aligned}
> \text{distributiveL} &:= x \&\text{and} (y \&\text{or} z) \\
\text{distributiveL} &:= (x \wedge (y \vee z))
\end{aligned} \tag{12.32}$$

$$\begin{aligned}
> \text{distributiveR} &:= (x \&\text{and} y) \&\text{or} (x \&\text{and} z) \\
\text{distributiveR} &:= (x \wedge y) \vee (x \wedge z)
\end{aligned} \tag{12.33}$$

To confirm the equivalence of the two Boolean expressions, we use the **Equivalent** command. This command requires two arguments, the two Boolean expressions that are to be tested for equivalence. The command returns true if the expressions are equivalent and false if not.

```
> Equivalent(distributiveL, distributiveR)
true
```

(12.34)

This verifies the given distributive law.

The **Equivalent** command also accepts an optional third argument. In the case that the two expressions are not equivalent, if you provide an unevaluated name (a name in single right quotes), then that name will be assigned to a set of assignments of truth values to the variables in the expression that demonstrate that the expressions are not equivalent.

Consider the nonequivalence: $x + xy \neq y$. In logical form, this is $x \vee (x \wedge y) \neq y$. Apply the **Equivalent** command with third argument '**P**'.

```
> Equivalent((x &or (x &and y)), y, 'P')
false
```

(12.35)

The name **P** now stores a set indicating assignments for x and y .

```
> P
{x = false, y = true}
```

(12.36)

This output means that setting x equal to false and y equal to true provides a demonstration, by counterexample, that $x \vee (x \wedge y) \neq y$. Indeed, substituting $x = \text{false}$ and $y = \text{true}$ on the left hand side produces

$$\text{false} \vee (\text{false} \wedge \text{true}) \equiv \text{false} \vee \text{false} \equiv \text{false}.$$

That is not the same as the right-hand side, y , which is assigned *true*.

Note that the single right quotes around the name in the third argument of **Equivalent** ensure that, should **P** have already stored a value, that value would be overwritten. If you omit the single quotes and **P** has a value already stored in it, an error will result.

Equality of Boolean Functions

Equality of Boolean functions can also be checked with the **Equivalent** command. You do this by applying the command to the two functions evaluated on the same variables.

Consider the following Boolean functions.

$$f_1(x, y) = \overline{(xy)}$$

$$f_2(x, y) = \bar{x} + \bar{y}.$$

Define the corresponding functional operators:

```
> f1 := (x, y) → BooleanSimplify(&not (x &and y))
f1 := (x, y) ↦ BooleanSimplify(¬(x ∧ y))
```

(12.37)

$$\begin{aligned}
 &> f2 := (x, y) \rightarrow \text{BooleanSimplify}((\&\text{not } x) \&\text{or } (\&\text{not } y)) \\
 &\quad f2 := (x, y) \mapsto \text{BooleanSimplify}((\neg x) \vee (\neg y))
 \end{aligned}
 \tag{12.38}$$

We can test the assertion that $f_1(x, y) = f_2(x, y)$ by applying the **Equivalent** command with arguments **f1(x,y)** and **f2(x,y)**.

$$\begin{aligned}
 &> \text{Equivalent}(f1(x, y), f2(x, y)) \\
 &\quad \text{true}
 \end{aligned}
 \tag{12.39}$$

Duality

We conclude this section by introducing the **Dual** command. This command accepts only one argument, a Boolean expression, and produces the dual of that expression.

For example, consider the expression $x \cdot \bar{y} + y \cdot \bar{z} + \bar{x} \cdot z$. As a logical expression, this can be written as $(x \wedge \neg y) \vee (y \wedge \neg z) \vee (\neg x \wedge z)$. We calculate the dual by applying the **Dual** command.

$$\begin{aligned}
 &> \text{Dual}((x \&\text{and } \&\text{not } (y)) \&\text{or } (y \&\text{and } \&\text{not } (z)) \&\text{or } (\&\text{not } (x) \&\text{and } z)) \\
 &\quad ((x \vee (\neg y)) \wedge (y \vee (\neg z))) \wedge ((\neg x) \vee z)
 \end{aligned}
 \tag{12.40}$$

Similarly, the dual of $\bar{x}y + \bar{y}z + x\bar{z}$ can be computed by

$$\begin{aligned}
 &> \text{Dual}((\&\text{not } (x) \&\text{and } y) \&\text{or } (\&\text{not } (y) \&\text{and } z) \&\text{or } (x \&\text{and } \&\text{not } (z))) \\
 &\quad (((\neg x) \vee y) \wedge ((\neg y) \vee z)) \wedge (x \vee (\neg z))
 \end{aligned}
 \tag{12.41}$$

Note that Exercise 13 of Section 12.1 asks you to prove that the expressions $x \cdot \bar{y} + y \cdot \bar{z} + \bar{x} \cdot z$ and $\bar{x} \cdot y + \bar{y} \cdot z + x \cdot \bar{z}$ are equivalent. The duality principle implies that the duals calculated above are also equivalent. This can be verified by the **Equivalent** command.

$$\begin{aligned}
 &> \text{Equivalent}((12.40), (12.41)) \\
 &\quad \text{true}
 \end{aligned}
 \tag{12.42}$$

12.2 Representing Boolean Functions

In this section, we will see how to use Maple to express Boolean functions in the disjunctive normal form (also called sum-of-products expansion). We will first look at the Maple command for turning an expression in Boolean algebra into the disjunctive normal form. Then, we will see how to write a procedure for finding an expression based on a table of values.

Disjunctive Normal Form from an Expression

Given an expression written in terms of the **Logic** operators, the **Canonicalize** command can be used to transform the expression into disjunctive normal form.

As an example, consider Example 3 of Section 12.2: $(x + y)\bar{z}$. In logical form, this is $(x \vee y) \wedge \neg z$. We assign this logical expression to a name.

$$\begin{aligned}
 &> \text{Example3} := (x \&\text{or } y) \&\text{and } \&\text{not } (z) \\
 &\quad \text{Example3} := (x \vee y) \wedge (\neg z)
 \end{aligned}
 \tag{12.43}$$

The **Canonicalize** command has several forms. To transform an expression into disjunctive normal form, two arguments are required. The first argument is the expression to be transformed. Second, you must provide a set or list containing the variables appearing in the expression.

$$\begin{aligned} &> \text{Canonicalize}(\text{Example3}, \{x, y, z\}) \\ &((\neg z) \wedge x \wedge y) \vee ((\neg z) \wedge x \wedge (\neg y)) \vee ((\neg z) \wedge y \wedge (\neg x)) \end{aligned} \quad (12.44)$$

Note that this agrees with the solution to Example 3.

The **Canonicalize** command also accepts a third argument used to specify the type of canonical form desired. The default behavior is to produce disjunctive normal form, but this can be emphasized by including the option **form=DNF**.

$$\begin{aligned} &> \text{Canonicalize}(\text{Example3}, \{x, y, z\}, \text{form} = \text{DNF}) \\ &((\neg z) \wedge x \wedge y) \vee ((\neg z) \wedge x \wedge (\neg y)) \vee ((\neg z) \wedge y \wedge (\neg x)) \end{aligned} \quad (12.45)$$

To produce conjunctive normal form instead, you use the **form=CNF** option.

$$\begin{aligned} &> \text{Canonicalize}(\text{Example3}, \{x, y, z\}, \text{form} = \text{CNF}) \\ &(x \vee y \vee z) \wedge (x \vee y \vee (\neg z)) \wedge (x \vee (\neg y) \vee (\neg z)) \wedge (y \vee (\neg x) \vee (\neg z)) \\ &\quad \wedge ((\neg x) \vee (\neg y) \vee (\neg z)) \end{aligned} \quad (12.46)$$

There is a third option, **form=MOD2**, which results in the 0–1 canonical form. However, as we mentioned earlier, Maple interprets + as the exclusive or, and thus the result of this option will be different from what is described in the textbook.

The **Normalize** command can also be used to produce a disjunctive normal form expression. Note that an argument specifying the variables is not used by **Normalize**. Like **Canonicalize**, the default is disjunctive normal form, but **form=DNF** and **form=CNF** are both accepted by **Normalize**.

Here is the result of **Normalize** applied to Example 3.

$$\begin{aligned} &> \text{Normalize}(\text{Example3}) \\ &((\neg z) \wedge x) \vee ((\neg z) \wedge y) \end{aligned} \quad (12.47)$$

Note that the result is not the same as before. In fact, if you compare this result to the solution of Example 3, you will see that this expression is equivalent to the second line in the step-by-step expansion in the solution. In particular, it is the result of applying the distributive law to the original expression.

Normalize produces an expression in disjunctive normal form. That is, the result is a disjunction of terms with each term consisting of a conjunction of variables and their negations. It does not, however, produce canonical disjunctive normal form, like **Canonicalize** does.

The benefit of **Normalize** is that, as you can see, it is typically simpler than canonical disjunctive normal form. The benefit of **Canonicalize** is that the canonical disjunctive normal form is unique. That is, two equivalent expressions necessarily have the same canonical disjunctive normal form, while **Normalize** may express them differently. While the textbook does not emphasize this, the examples and techniques it describes are for producing the canonical sum-of-products expansion.

Disjunctive Normal Form from a Table

Example 1 of Section 12.2 describes how to find an expression for a Boolean function represented by a table of values. Here, we will show how to write a procedure to accomplish this task.

First, we must decide how we will represent the table of values in Maple. Rather than creating a Maple table object, we will represent the table as the set of those assignments of truth values to variables for which the function returns true.

For example, consider the function defined by the following table.

x	y	z	$F(x, y, z)$
<i>true</i>	<i>true</i>	<i>true</i>	<i>false</i>
<i>true</i>	<i>true</i>	<i>false</i>	<i>true</i>
<i>true</i>	<i>false</i>	<i>true</i>	<i>false</i>
<i>true</i>	<i>false</i>	<i>false</i>	<i>true</i>
<i>false</i>	<i>true</i>	<i>true</i>	<i>false</i>
<i>false</i>	<i>true</i>	<i>false</i>	<i>true</i>
<i>false</i>	<i>false</i>	<i>true</i>	<i>true</i>
<i>false</i>	<i>false</i>	<i>false</i>	<i>false</i>

There are four rows in the table for which the function returns true. We represent this table by forming the set consisting of the four lists of values for x , y , and z corresponding to those rows.

```
> exampleTable := {[false,false,true],[false,true,false],  
  [true,false,false],[true,true,false]}  
exampleTable := {[false,false,true],[false,true,false],  
  [true,false,false],[true,true,false]} (12.48)
```

This will be the first argument to the procedure we create. The procedure will also need names for the variables; therefore, we also require a list of names. This will be the second argument to the procedure.

```
> exampleVariables := [x,y,z]  
exampleVariables := [x,y,z] (12.49)
```

To form the Boolean expression that represents this function, we follow Example 1. For each row in the table for which the function returns true, that is, for each element in **exampleTable**, we produce the corresponding minterm.

To create a minterm associated with an element of **exampleTable**, we proceed as follows. Initialize a local variable to **NULL**. This variable will hold the sequence of variables or their negations. Then, begin a for loop with loop variable ranging from 1 to the number of variables. Within the loop, test whether or not the entry in the list of truth values is true or not. If the entry is true, then update the sequence by adding the name of the corresponding variable. If the entry is false, then update the sequence by appending the negation of the variable. Finally, apply **&and** to the sequence of variables or negations of variables. This approach, rather than applying **&and** at each step, will result in minterms without extra parentheses.

The following procedure accepts a single list of truth values (a single row) and the list of variables, and produces the minterm.

```

1 FormMinterm := proc (row : list (truefalse) , vars : list (symbol) )
2     local mintermSeq, i;
3     uses Logic;
4     if nops (row) <> nops (vars) then
5         error " Incorrect number of variables ";
6     end if ;
7     mintermSeq := NULL;
8     for i from 1 to nops (row) do
9         if row[i] then
10            mintermSeq := mintermSeq, vars[i];
11        else
12            mintermSeq := mintermSeq, &not (vars[i]);
13        end if ;
14    end do;
15    return &and (mintermSeq) ;
16 end proc:

```

This procedure, applied to a member of the **exampleTable**, produces the corresponding minterm.

> *FormMinterm* ([true, true, false], [x, y, z])
 $x \wedge y \wedge (\neg z)$ (12.50)

To create the Boolean expression for the function, all that remains is to form the disjunction of the minterms produced by **FormMinterm**. The procedure below accepts the table representation and list of variables as arguments. It first checks to see if it was passed the empty set, and if so, returns the expression **false**. Similar to **FormMinterm**, it initializes a sequence to **NULL**. For each element of the table representation, it calls **FormMinterm** and adds the output from that procedure to the sequence. At the end, the **&or** operator is applied to the sequence of minterms.

```

1 BooleanFromTable := proc (T : set (list (truefalse)) , V : list (symbol) )
2     local mtSeq, row, mt;
3     uses Logic;
4     if T = {} then
5         return false;
6     end if ;
7     mtSeq := NULL;
8     for row in T do
9         mt := FormMinterm (row, V) ;
10        mtSeq := mtSeq, mt;
11    end do;
12    return &or (mtSeq) ;
13 end proc:

```

Note that the structured type **set(list(truefalse))** ensures that the first argument is a **set** whose members are each a **list** with members of type **truefalse** (namely **true** or **false**).

Applying the procedure to our example table produces the desired Boolean expression.

> *BooleanFromTable(exampleTable, exampleVariables)*

$$((\neg x) \wedge (\neg y) \wedge z) \vee ((\neg x) \wedge y \wedge (\neg z)) \vee (x \wedge (\neg y) \wedge (\neg z)) \\ \vee (x \wedge y \wedge (\neg z))$$


(12.51)

12.3 Logic Gates

In this section, we will use Maple to work with logic gates, particularly circuit diagrams. First, we will see how to use Maple to translate a circuit diagram into an expression using the **Logic** operators. Then, we will do the reverse and see how to transform a logical expression into a circuit diagram (modeled as a tree diagram).

Circuit Diagram to Logical Expression

Consider the circuit diagram shown below.



Our goal in this subsection is to use Maple to produce a logical expression for the output of this diagram.


To do this, we use the fact that the inert forms of the logical operators can be used in functional form. In particular, **&and** and **&or** can be used as functions applied to sequences of expressions, and **¬** can be applied to a single argument. For example, the following forms the disjunction of **x**, **y**, and **z**.

> **&or**(*x*, *y*, *z*)

$$x \vee y \vee z$$

(12.52)

We give each gate in the diagram a label. The specific names are not important. We chose to label the gates using the capital letter G with subscripts numbered from the right to the left.



Interpret the labels as names for both the gates themselves and for their outputs. Note that G_1 is the name for both the output of the final gate and also names the output of the circuit. First, we remove any existing assignments with **unassign**.

```
> unassign('G1', 'G2', 'G3', 'G4', 'G5', 'a', 'b', 'c')
```

The input of G_1 is the outputs from gates G_2 and G_3 . That is to say, $G_1 = G_2 \text{ OR } G_3$. We can write that in Maple.

```
> G1 := &or(G2, G3)
      G1 := G2 ∨ G3
```

(12.53)

For each gate, do the same. Note that the order in which the gates are specified is irrelevant. The output G_2 is $a \text{ AND } b$.

```
> G2 := &and(a, b)
      G2 := a ∧ b
```

(12.54)

The output of G_3 is the conjunction of G_4 , b , and G_5 .

```
> G3 := &and(G4, b, G5)
      G3 := G4 ∧ b ∧ G5
```

(12.55)

And G_4 and G_5 are the results of inversion on a and c , respectively.

```
> G4 := &not(a)
      G4 := ¬a
```

(12.56)

```
> G5 := &not(c)
      G5 := ¬c
```

(12.57)

Once all of the gates have been specified, inspect the value for the final gate, G_1 .

```
> G1
      (a ∧ b) ∨ ((¬a) ∧ b ∧ (¬c))
```

(12.58)

This tells us that the circuit's result in 0–1 form is $ab + \bar{a}b\bar{c}$.

The reason this works is that when we define the output of a gate in terms of unassigned names, such as **G2**, Maple accepts the definition. When **G2** is later assigned its own value and then the statement **G1** is executed, Maple resolves all assigned names into their definitions so that the expression for **G1** is in terms of unassigned names (**a**, **b**, and **c**) only.

Logical Expression to Circuit Diagram

We have just seen how to use Maple to transform a circuit diagram into a logical expression for the result of the circuit. Now we consider the reverse. Given a logical expression, such as that for **G1**, we will use Maple to transform the expression into a circuit diagram.

We will model a circuit diagram as an ordered rooted tree. While circuit diagrams are not necessarily ordered, making the tree ordered will allow for the possibility of including Boolean functions as subcircuits.

Vertices in the tree will correspond to gates in the circuit. One of the vertices is distinguished as the root, which will correspond to the final gate, whose output is the output of the circuit. Each vertex has a number of children vertices. The edge between a vertex and its child corresponds to the input to the gate. Each vertex other than the root has a parent, and the edge from the vertex to the parent corresponds to the output from the gate.

The assumption that a circuit can be modeled as a tree requires that the circuit satisfy the following properties. First, the circuit has only one output. Second, each gate has only one output. Third, there are no branches (e.g., an input cannot be used as input to more than one gate). Given that our goal is to begin with a logical expression and create a corresponding abstract circuit, these restrictions are of little importance. If we were actually building physical circuits, there would be efficiency concerns.

Recall that in Chapter 11, we wrote the procedure **InfixToTree** for converting an algebraic expression in terms of inert versions of the binary arithmetic operators into a tree representation. We will make use of the procedures from Chapter 11 here, and have included them in the package for this chapter as well. Thus, if you apply **with** to either “Chapter 11” or “Chapter 12,” the needed commands will be included.

Recall that the procedures written in each chapter of this manual are collected in packages. If you have not moved the library provided with the manual into Maple’s default library directory (you can obtain the directory by inspecting the value of **libname**), then you should tell Maple where to look for the manual’s library by executing a command like one of the two below.

```
> libname := libname, "/Users/danieljordan/DiscreteMath/"  
> libname := libname, FileTools[ParentDirectory](  
    Maplets[Utilities][GetFile]('title' = "Locate RosenMaplePackages.mla",  
    'directory' = currentdir(homedir), filefilter = "mla",  
    filterdescription = "Library Files"))
```

For the first of the commands above, you would need to replace the given directory path with a string appropriate to your system and location of the provided .mla file. The second will open a dialog that you can use to navigate to the directory. Note that these commands have been made nonexecutable so that they will not create errors or open dialogs if you decide to execute the entire worksheet. Once Maple is able to find the library file for the manual, the following will load the needed procedures and type definitions.

```
> with(Chapter12, VOrderComp, DrawORTree, NewExpressionTree,  
    JoinTrees, InfixToTree) :
```

Please refer to the Introduction or Maple’s documentation if you need more instructions on how to use the packages on your system.


Observe what happens if we apply the **InfixToTree** command to the logical expression we obtained from the circuit diagram above.

> *InfixToTree (G1)*

Graph 1: a directed unweighted graph with 10 vertices and 9 arc(s)

(12.59)

> *DrawORTree (%)*



Observe that the original circuit diagram and the tree have the same structure. After reversing the arrows, rotating by 90° , and exchanging the symbols with the inert commands, the two are identical.

12.4 Minimization of Circuits

In this section, we will discuss the use of the **BooleanSimplify** command for minimizing circuits. Then we will create a brute force algorithm for handling don't care conditions. Finally, we will provide an implementation of the Quine–McCluskey method.

The BooleanSimplify Command

We described the **BooleanSimplify** command in the first section of this chapter. The command accepts only one argument, a Boolean expression.

For example, we apply **BooleanSimplify** to G1, the expression we obtained for the output of the circuit diagram at the beginning of Section 12.3.

> *BooleanSimplify (G1)*

$(a \wedge b) \vee (b \wedge (\neg c))$

(12.60)

The result indicates that $\neg a$ can be removed as an input to the second AND gate.

Note that the result of **BooleanSimplify** is guaranteed to be minimal. That is, it is not possible to reduce it further. It is not, however, guaranteed that the result is a minimum sum of prime

implicants. That is, while no simplification of the output from **BooleanSimplify** is possible, it may be the case that there is a simpler expression equivalent to the original input.

The reason for this lies in the final step of the Quine–McCluskey method. Once the essential prime implicants have been identified, you are left with prime implicants that are not essential and you must identify the best choice of those prime implicants that will complete the cover. To find the minimum expression, you use a backtracking approach (a depth-first search). Unfortunately, this requires exponential time.

The alternative is to use a heuristic approach, choosing the prime implicants that cover the most minterms. This is considerably more efficient, but does not guarantee that the resulting expression is the minimum. At the conclusion of this section, we will design an implementation of the Quine–McCluskey method using such a heuristic approach. Implementing a backtracking approach is left as an exercise.

Don't Care Conditions

Informally, a set of don't care conditions for a Boolean function F is a set of points in the domain of F whose images do not concern us.

If F is a function on n variables, then its domain is $\{true, false\}^n$. Let A be the subset of $\{true, false\}^n$ for which the values of F are specified. If we think of F as fully defined on this subset A , then we are interested in the family of all extensions of F to all of $\{true, false\}^n$. In other words, the set of all G defined on $\{true, false\}^n$ that agree with F on A . The goal is to choose the particular G that is “simplest.” That is, the G that has the smallest sum of products expansion.

We should pause to consider the size of this problem. If there are d don't care points, then there are 2^d possible extensions G . Considering every possible extension can become rather time consuming.

The procedure we write will make use of the **BooleanFromTable** procedure from Section 12.2. Recall that the **BooleanFromTable** procedure accepted a set consisting of those points for which the function returns true. The points are represented by lists of trues and falses. It also required a list of the names of the variables. **BooleanFromTable** returns the conjunctive normal form of the function that returns true on the specified points and false on all others.

In **DontCare**, we will loop through every possible extension G of the input function F . Specifically, the procedure will accept two sets of points. One representing those points for which the function must return true, and the second set of points representing the don't care conditions. It is understood that the function must return false on all points in neither set. **DontCare** will also accept a list of variable names.

Each extension G corresponds to a subset of the don't care conditions. We will use the **subsets** command, described in Section 6.1 of this manual. Recall that **subsets**, when applied to a set, returns a table with two elements. The **finished** element is a Boolean value set to true once all of the subsets of the given set have been listed. The **nextvalue** entry is a procedure that, when executed, produces the next subset.

For each subset of the don't care conditions, **DontCare** will apply **BooleanFromTable** to the union of the subset and the set of points for which F must be true. It will then apply **BooleanSimplify** to the result.

To determine the minimal expression, we compare expressions using the **length** command, which returns an integer representing the length of the expression. (This is a crude comparison, but it will suffice.) We use the standard approach of storing the simplest expression found so far, and replacing it each time a shorter expression is located. Note that the first set produced by **subsets** is always the empty set, so we initialize the temporary minimum to the function in which all don't care conditions are taken to be false.

```

1 DontCare := proc (T :: set (list (truefalse)), DC :: set (list (truefalse)),
2   V :: list (symbol))
3   local minExpr, minLength, S, s, nextExpr;
4   uses Logic;
5   S := combinat[subsets] (DC);
6   s := S[nextvalue] ();
7   minExpr := BooleanSimplify (BooleanFromTable (T, V));
8   minLength := length (minExpr);
9   while not S[finished] do
10      s := S[nextvalue] ();
11      nextExpr := BooleanSimplify (BooleanFromTable (T union s, V));
12      if length (nextExpr) < minLength then
13         minExpr := nextExpr;
14         minLength := length (nextExpr);
15      end if;
16   end do;
17   return minExpr;
18 end proc;

```

Consider the Boolean function F defined by the following table of values, in which “d” in the final column indicates a don't care condition.

x	y	z	$F(x, y, z)$
false	false	false	true
false	false	true	false
false	true	false	d
false	true	true	d
true	false	false	true
true	false	true	false
true	true	false	false
true	true	true	true

The points that must evaluate to true are

$\{[false, false, false], [true, false, false], [true, true, true]\},$

and the don't care conditions are

$\{[false, true, false], [false, true, true]\}.$

We apply the procedure **DontCare** with these two sets as inputs along with the list of variables.

```
> DontCare ({[false,false,false],[true,false,false],[true,true,true]},
            {[false,true,false],[false,true,true]},['x','y','z'])
(y ∧ z) ∨ ((¬y) ∧ (¬z))
```

(12.61)

Before leaving don't care conditions, we should mention that the Quine–McCluskey method, which is the subject of the next subsection, provides a much more efficient solution than the procedure **DontCare**.

To take don't care conditions into account with the Quine–McCluskey method, you include them in the list of minterms that are used to generate prime implicants, but you do not include them in the list of minterms that need to be covered by the prime implicants. In terms of Example 9 of the text, don't care conditions appear in the first column of Table 3, but are omitted from the top row of Table 4.

Quine–McCluskey Method

We conclude with an implementation of the Quine–McCluskey method. This method is fairly involved and it will take considerable effort to implement it correctly.

It will be helpful to have an example that we can use to illustrate the method as we build the procedure. The expression we use for the example is

$$wxyz + w\bar{x}yz + w\bar{x}y\bar{z} + w\bar{x}y\bar{z} + \bar{w}xyz + \bar{w}x\bar{y}z + \bar{w}x\bar{y}\bar{z} + \bar{w}x\bar{y}\bar{z} + \bar{w}x\bar{y}z + \bar{w}x\bar{y}z.$$

We assign this to the name **F**.

```
> F := (w &and x &and (&not y) &and (&not z)) &or
(w &and (&not x) &and y &and z) &or
(w &and (&not x) &and y &and (&not z)) &or
(w &and (&not x) &and (&not y) &and (&not z)) &or
(&not w &and x &and y &and z) &or
(&not w &and x &and (&not y) &and z) &or
(&not w &and x &and (&not y) &and (&not z)) &or
(&not w &and (&not x) &and y &and z) &or
(&not w &and (&not x) &and (&not y) &and z) &or
(&not w &and (&not x) &and (&not y) &and (&not z)) :
```

Let us begin by (very) briefly outlining the approach. More details will be given as we proceed.

1. Transform the minterms into bit strings.
2. Group the bit strings by the number of 1s.
3. Combine bit strings that differ in exactly one location.
4. Repeat steps 2 and 3 until no additional combinations are possible.
5. Identify the prime implicants (those bit strings not involved in a simplification) and form the coverage table.
6. Identify the essential prime implicants and update the table.
7. Process the remaining prime implicants using a heuristic approach in order to achieve complete coverage.

Implementing this will require several different procedures that will come together to achieve the goal of minimizing the expression for F .

Modifying Arguments

Before we begin implementing the method, we take a moment to reiterate the use of **evaln** as a parameter modifier, which we first described in Section 11.2. This will be used later in the implementation to avoid the need to copy data structures that must be modified by a procedure.

Modifying a parameter with **evaln** means that, instead of sending a copy of the object to be worked on by the procedure, you send the name of the object. This allows the argument to be modified directly within the procedure. However, every time you want to access the value of the object, rather than the name, you must apply **eval** to the parameter.

The procedure below adds 3 to its argument, changing the value stored in the name it is passed.

```

1 | add3 := proc (x : evaln (integer) )
2 |     x := eval (x) + 3;
3 | end proc:

```

Applying **add3** to a name that stores an integer will now alter the value stored in the name.

> *two* := 2
two := 2 (12.62)

> *add3*(*two*)
5 (12.63)

> *two*
5 (12.64)

Transforming Minterms into Bit Strings

The first task is to process the input. That is, F must be transformed into a list of bit strings. This is not strictly necessary, but it makes working with the minterms more convenient. We represent bit strings as lists of 0s and 1s.

We begin by creating a procedure to transform a single minterm into a bit string. We assume that the input to this procedure will be a properly formed minterm, that is, a conjunction of variables and negations of variables. We require that a list of variables be provided to the procedure, so that the bit string can be formed in the proper order.

Consider the following minterm, which is the fourth minterm in our example F .

> *minterm* := *w* &and ¬(*x*) &and *y* &and ¬(*z*)
minterm := ((*w* ∧ (¬*x*)) ∧ *y*) ∧ (¬*z*) (12.65)

In order to transform this into the bit string [1, 0, 1, 0], we must first determine the variables and negations of variables that are conjoined. Our first goal, therefore, is to transform the minterm into the list [*w*, **not** *x*, *y*, **not** *z*].

For this, recall that the **op** command can be used to extract the operands of an expression. In this case, **op** will return the two operands of the final **&and**.

$$\begin{aligned} &> [op(minterm)] \\ &\quad [(w \wedge (\neg x)) \wedge y, \neg z] \end{aligned} \tag{12.66}$$

In order to obtain a list of the conjoined variables and negations, we will repeatedly apply **op**.

We begin by initializing a list consisting of **minterm** as the only object and we set an index variable equal to 1.

$$\begin{aligned} &> MTList := [minterm] \\ &\quad MTList := [((w \wedge (\neg x)) \wedge y) \wedge (\neg z)] \end{aligned} \tag{12.67}$$

$$\begin{aligned} &> i := 1 \\ &\quad i := 1 \end{aligned} \tag{12.68}$$

We create a while loop that will continue as long as the index variable is not greater than the length of **MTList**. Within the loop, we consider **MTList[i]**. Comparing **op(0,MTList[i])** to the operator '**&and**' will tell us whether or not the *i*th member of the list is a conjunction. If so, we apply **op** to it, using **subsop** to replace the **i** location in the list with **op(MTList[i])**. On the other hand, if **op(0,MTList[i])** is not '**&and**', then we increment **i**.

$$\begin{aligned} &> \text{while } i \leq nops(MTList) \text{ do} \\ &\quad \text{if } op(0, MTList[i]) = \&and \text{ then} \\ &\quad \quad MTList := subsop(i = op(MTList[i]), MTList) \\ &\quad \text{else} \\ &\quad \quad i := i + 1 \\ &\quad \text{end if} \\ &\text{end do} \end{aligned}$$

This has transformed **MTList** into a list of the conjoined variables and negations of variables.

$$\begin{aligned} &> MTList \\ &\quad [w, \neg x, y, \neg z] \end{aligned} \tag{12.69}$$

To complete the transformation into a bit string, we only need to check, for each variable, whether the variable or its negation is in the list. Recall that we will insist that the procedure be given the list of variables as an argument to maintain the proper order of the variables.

We first assign the list of variables to a name.

$$\begin{aligned} &> variableList := [w, x, y, z] \\ &\quad variableList := [w, x, y, z] \end{aligned} \tag{12.70}$$

Now create a list, initialized to the proper length, for the bit string.

$$\begin{aligned} &> Bitstring := [0 \$ nops(variableList)] \\ &\quad Bitstring := [0, 0, 0, 0] \end{aligned} \tag{12.71}$$

Finally, we use a for loop to check, for each variable, whether the variable is in **MTList**. If the variable is a member of **MTList**, then we change the bit to 1. Otherwise, we assume that the negation is in the minterm and we leave the value in the bit string as 0.

```
> for i from 1 to nops(variableList) do
    if variableList[i] in MTList then
        Bitstring[i] := 1
    end if
end do
```

This has created the bit string associated to **minterm**.

```
> Bitstring
[1, 0, 1, 0] (12.72)
```

We condense this process into a single procedure.

```
1 MTtoBitString := proc(minterm, variableList : list(symbol))
2     local MTList, i, Bitstring;
3     uses Logic;
4     MTList := [minterm];
5     i := 1;
6     while i <= nops(MTList) do
7         if op(0, MTList[i]) = '&and' then
8             MTList := subsop(i=op(MTList[i]), MTList);
9         else
10            i := i + 1;
11        end if;
12    end do;
13    Bitstring := [0 $ nops(variableList)];
14    for i from 1 to nops(variableList) do
15        if variableList[i] in MTList then
16            Bitstring[i] := 1;
17        elif '&not'(variableList[i]) in MTList then
18            Bitstring[i] := 0;
19        else
20            error "Unrecognized object in MTList.";
21        end if;
22    end do;
23    return Bitstring;
24 end proc;
```

```
> MTtoBitString(minterm, [w, x, y, z])
[1, 0, 1, 0] (12.73)
```

Transforming the Original Expression into Bit Strings

Now that we have the means for transforming a single minterm into a bit string, we are ready to transform an expression in disjunctive normal form into a list of bit strings.

This works in nearly the same way as **MTtoBitString** did. Given an expression in disjunctive normal form, we break it into a list, **DNFList**, but this time, instead of looking for the operator to be **'&and'**, it must be **'&or'**. Once the list is formed, we apply **MTtoBitString** on each element of the list.

Here is the procedure. Notice that the first while loop is very similar to **MTtoBitString**, but after the while loop, we complete the procedure with an application of **map**. We use **variableList** as a third argument to **map**. When **map** is given more than two arguments, subsequent arguments are treated as additional arguments to the procedure. That is, **map(P,L,a)** applies the procedure **P** to **(L,a)** for each **L** in the list **L**. In this case, **MTtoBitString** requires that the list of variables be passed to it each time it is called.

```

1  DNFToBitList := proc(dnfExpr, variableList : :list(symbol) )
2      local DNFList, i;
3      uses Logic;
4      DNFList := [dnfExpr];
5      i := 1;
6      while i <= nops(DNFList) do
7          if op(0, DNFList[i]) = '&or' then
8              DNFList := subsop(i=op(DNFList[i]), DNFList);
9          else
10             i := i + 1;
11         end if;
12     end do;
13     return map(MTtoBitString, DNFList, variableList);
14 end proc;
```

Apply this procedure to the example expression.

$$\begin{aligned}
 &> \text{Fbits} := \text{DNFToBitList}(F, [w, x, y, z]) \\
 &\quad \text{Fbits} := [[1, 1, 0, 0], [1, 0, 1, 1], [1, 0, 1, 0], [1, 0, 0, 0], [0, 1, 1, 1], \\
 &\quad \quad [0, 1, 0, 1], [0, 1, 0, 0], [0, 0, 1, 1], [0, 0, 0, 1], [0, 0, 0, 0]]
 \end{aligned} \tag{12.74}$$

Transforming Bit Strings into Minterms

At the conclusion of the Quine–McCluskey process, we will want to display the result in disjunctive normal form. This will require that we turn bit strings back into minterms.

Note that since this procedure will be applied at the end of the process, it may be that some of the variables have been removed. We will be using the string “-” in a bit string to indicate the elimination of a variable.

This procedure will require the bit string and a list of variable names as its input. It operates in two stages. First, it processes the variable list based on the content of the bit string. We make a copy of the variable list and then modify it by applying **¬** when the entry in the bit string is 0 and by replacing the variable with “-” when that is the entry in the bit string.

$$\begin{aligned}
 &> \text{varList} := [w, x, y, z] \\
 &\quad \text{varList} := [w, x, y, z]
 \end{aligned} \tag{12.75}$$

$$\begin{aligned}
 &> \text{bitstr} := [0, 1, "-", 0] \\
 &\quad \text{bitstr} := [0, 1, "-", 0]
 \end{aligned} \tag{12.76}$$

```

> for i from 1 to nops(varList) do
  if bitstr[i] = 0 then
    varList[i] := &not(varList[i])
  elif bitstr[i] = "-" then
    varList[i] := "-"
  end if
end do

```

```

> varList
[¬w, x, "-", ¬z]

```

(12.77)

Once this processing has been done, we remove any occurrences of “-” by replacing them with **NULL**. The **subs** command accepts as its arguments a sequence of equations and a final expression. For example, **subs(x=a,expr)**. The result of this statement is that every occurrence of **x** in **expr** is replaced by **a**. With the equation “-”=**NULL** and the processed variable list as the arguments, **subs** will return the list with each “-” removed.

```

> varList := subs("-", NULL, varList)
varList := [¬w, x, ¬z]

```

(12.78)

To form the conjunction, we simply apply **op** to extract the sequence of variables and negations from the enclosing list and then apply **&and**.

```

> &and(op(varList))
(¬w) ∧ x ∧ (¬z)

```

(12.79)

We combine these steps into a procedure.

```

1 BitStringtoMT := proc(bitstring, variableList : list(symbol))
2   local varList, i;
3   uses Logic;
4   varList := variableList;
5   for i from 1 to nops(varList) do
6     if bitstring[i] = 0 then
7       varList[i] := &not(varList[i]);
8     elif bitstring[i] = "-" then
9       varList[i] := "-";
10    end if;
11  end do;
12  varList := subs("-", NULL, varList);
13  return &and(op(varList));
14 end proc:

```

Applied to [0, 1, 0, 1] and [x, y, z, w], we see that **BitStringtoMT** reproduces the original minterm.

```

> BitStringtoMT([0, 1, 0, 1], [w, x, y, z])
(¬w) ∧ x ∧ (¬y) ∧ z

```

(12.80)

And applied to $[0, 1, \text{“} - \text{”}, 1]$, it removes the y .

$$\begin{aligned} &> \text{BitStringtoMT}([0, 1, \text{“} - \text{”}, 1], [w, x, y, z]) \\ &\quad (\neg w) \wedge x \wedge z \end{aligned} \tag{12.81}$$

Initializing the Source Table

In order to form the coverage table in the second part of the method, we need to know which of the original minterms are covered by which of the prime implicants. Refer to Tables 3 and 6 in the text. Notice that each bit string in those tables is associated with either a single number, in the case of the original minterms, or lists of numbers, for the derived products.

We will store this information in a table whose indices are the bit strings and whose entries are sets of integers. Given the **Fbits** list, we initialize this table with the elements of **Fbits** as the indices. The corresponding entries will be the set consisting of the bit string’s position in **Fbits**.

We will refer to this as the “coverage dictionary,” since it allows us to look up any bit string and determine all of the original minterms covered by it. The following procedure accepts the **Fbits** list as an argument and returns the coverage dictionary.

```

1  initCoverDict := proc (L : list)
2      local coverDict, i;
3      coverDict := table();
4      for i from 1 to nops(L) do
5          coverDict[L[i]] := {i};
6      end do;
7      return coverDict;
8  end proc;
```

Applying this procedure to **Fbits** produces the initial coverage dictionary. In order to inspect the entries, we must apply **eval** to the name of the table.

$$\begin{aligned} &> \text{coverageDict} := \text{initCoverDict}(\text{Fbits}) \\ &\quad \text{coverageDict} := \text{coverDict} \end{aligned} \tag{12.82}$$

$$\begin{aligned} &> \text{eval}(\text{coverageDict}) \\ &\quad \text{table}([[0, 1, 0, 0] = \{7\}, [0, 0, 1, 1] = \{8\}, [0, 1, 0, 1] = \{6\}, \\ &\quad \quad [0, 1, 1, 1] = \{5\}, [1, 1, 0, 0] = \{1\}, [0, 0, 0, 1] = \{9\}, [1, 0, 1, 0] = \{3\}, \\ &\quad \quad [1, 0, 0, 0] = \{4\}, [0, 0, 0, 0] = \{10\}, [1, 0, 1, 1] = \{2\}]) \end{aligned} \tag{12.83}$$

Grouping by the Number of 1s

Step 2 in our outline is to group the bit strings by the number of 1s.

The reason for this step is to improve the efficiency of finding simplifications to make. Since two bit strings can be combined only when they are identical except for one location, the only possible combinations are when one bit string has n 1s and the other has $n - 1$.

After step 1 is concluded, we have a list of bit strings. That will be the starting point for the procedure we create for this step. The result of this step will be to turn the list of bit strings into a table

of sets of bit strings, which we call **groups**. Associated to index i in **groups** will be the set of all bit strings with i 1s.

For each member of **Fbits**, we need to count the number of 1s. The **Occurrences** function from the **ListTools** package can do this for us, as illustrated below. We will use **ListTools** fairly extensively, so we load it to use the short forms of its commands.

```
> with(ListTools) :
> Occurrences(1, [1, 0, 1, 1, 0, 0, 1])
4
```

(12.84)

Observe that the first argument is the object being sought and the second is a list. The result is the number of occurrences of the first argument in the list. It is tempting to use the **add** command to add the bits in the list. We cannot do this, however, because after the first simplification, our bit strings will contain symbols that are not 1s or 0s.

To sort the members of **Fbits** into groups, based on the number of 1s, we will apply the **Classify** command, also in the **ListTools** package. This command requires as its first argument a function and a list as its second argument. It returns a **table** whose indices are the unique results of the function applied to the elements of the list, and, associated with each of those values is the set of members of the list which produce that value. For example, the following classifies a set of values based on the floor function.

```
> Classify(floor, [3, 5, 3.8, 6.2, 6.9, 3.02, 4.7, 5.5,  $\pi$ ])
table([3 = {3, 3.02, 3.8,  $\pi$ }, 4 = {4.7}, 5 = {5, 5.5}, 6 = {6.2, 6.9}])
```

(12.85)

Here is the result of classifying **Fbits**.

```
> groups := Classify(bitstr → Occurrences(1, bitstr), Fbits)
groups := table([0 = {[0, 0, 0, 0]}, 1 = {[0, 0, 0, 1], [0, 1, 0, 0], [1, 0, 0, 0]},
2 = {[0, 0, 1, 1], [0, 1, 0, 1], [1, 0, 1, 0], [1, 1, 0, 0]},
3 = {[0, 1, 1, 1], [1, 0, 1, 1]}])
```

(12.86)

In practice, we will want to ensure that every possible index from 0 to the length of a bit string appears in the table. One way to do this is by using the **tablemerge** function applied to the table in which every possible index is associated with the empty set. When given two tables as the only arguments, the second table's entry is used when an index appears in both. For example, the following would add indices 4 and 5 to the **groups** table, although nothing is being modified by this command.

```
> tablemerge(table([seq(j = {}, j = 0..5)]), groups)
table([0 = {[0, 0, 0, 0]}, 1 = {[0, 0, 0, 1], [0, 1, 0, 0], [1, 0, 0, 0]},
2 = {[0, 0, 1, 1], [0, 1, 0, 1], [1, 0, 1, 0], [1, 1, 0, 0]},
3 = {[0, 1, 1, 1], [1, 0, 1, 1]}, 4 = {}, 5 = {}])
```

(12.87)

We encapsulate this in a small procedure.

1	sortGroups := proc (bitstringList)
2	local groupTable, max, j;

```

3  uses ListTools;
4  groupTable := Classify (bitstr -> Occurrences (1, bitstr),
   bitstringList);
5  max := nops (bitstringList [1]);
6  return tablemerge (table ([seq (j={ }, j=0..max) ]), groupTable);
7  end proc;

```

Combining Bit Strings

Step 3 is to combine all of the bit strings that differ in exactly one location. We first write a procedure that takes as input two bit strings and either combines them if, in fact, they do differ in exactly one location, or returns false if they do not.

This procedure needs to do two tasks. First, it has to check to see whether or not the two bit strings differ in more than one location. Second, it needs to combine them if they are allowed to be combined.

Combining two bit strings is easy, provided we know the one location in which they differ. For example,

```

> bit1 := [1, "-", 0, 1, 1]
   bit1 := [1, "-", 0, 1, 1]

```

(12.88)

```

> bit2 := [1, "-", 0, 0, 1]
   bit2 := [1, "-", 0, 0, 1]

```

(12.89)

You can see that these are identical except in position 4.

To merge them, we take either one and replace position 4 with "-".

```

> subsop (4 = "-", bit1)
   [1, "-", 0, "-", 1]

```

(12.90)

We determine that they differ only in position 4 as follows. Begin by initializing a name **pos**, for position, to 0. This will hold the position at which the difference occurs. Setting it to 0 indicates that we have not found a difference.

Now use a for loop to compare each pair of entries in **bit1** and **bit2**. If we find a difference, check the value of **pos**. If **pos** is 0, then we know that this is the first time a difference was found and we set **pos** to the position of the difference. If **pos** is not 0, however, then we know that this is the second time a difference was found. In this case, the bit strings cannot be merged and we return false. If the loop completes without having returned false, then the two bit strings can be merged at position **pos**.

Here is the procedure.

```

1  MergeBitstrings := proc (bit1 : list, bit2 : list)
2      local i, pos;
3      pos := 0;
4      for i from 1 to nops (bit1) do
5          if bit1 [i] <> bit2 [i] then

```

```

6         if pos = 0 then
7             pos := i;
8         else
9             return false;
10        end if;
11    end if;
12 end do;
13 return subsop(pos="-", bit1);
14 end proc;

```

We see that it works correctly on our two example bit strings.

```

> MergeBitstrings(bit1, bit2)
[1, "-", 0, "-", 1]

```

(12.91)

Searching for Combinations to Make

The **MergeBitstrings** procedure will do the work of checking to see if bit strings can be merged and returning the result if they can. However, we need to give **MergeBitstrings** the bit strings to test.

Recall that, in our example, we have successfully grouped the minterms by the number of 1s they contain.

```

> eval(groups)
table ([0 = {[0, 0, 0, 0]}, 1 = {[0, 0, 0, 1], [0, 1, 0, 0], [1, 0, 0, 0]},
      2 = {[0, 0, 1, 1], [0, 1, 0, 1], [1, 0, 1, 0], [1, 1, 0, 0]},
      3 = {[0, 1, 1, 1], [1, 0, 1, 1]})

```

(12.92)

Here, we will produce a list containing all the bit strings formed by merging two members of **groups**. Since there may be multiple ways to obtain the same bit string, we store these as a set. We initialize to the empty set.

```

> Fbits1 := {}
Fbits1 := ∅

```

(12.93)

Also recall that it is only possible to merge bit strings when one has n 1s and one has $n - 1$ 1s. This suggests a for loop with n ranging from 1 to the maximum index in **groups**. Within the body of the for loop, we will consider the sets with $n - 1$ 1s and the set with n 1s.

The loop is structured as follows:

```

> for n to max(indices(groups)) do
    A := groups[n];
    B := groups[n - 1]
end do :

```

After A and B have been defined, we need to compare every possible pair. We use two more for loops, one for each member of A and one for each member of B . Within the inner for loop, we use **MergeBitstrings** and store the result. If it is not false, we add it to the new list of bit strings, **Fbits1**.

```

> for n from 1 to max(indices(groups)) do
  A := groups[n];
  B := groups[n - 1]
  for a in A do
    for b in B do
      m := MergeBitstrings(a, b);
      if m ≠ false then
        Fbits1 := Fbits1 union {m};
      end if
    end do
  end do
end do :

> Fbits1 := [op(Fbits1)]
Fbits1 := [[0, 0, 0, “-”], [0, 0, “-”, 1], [0, 1, 0, “-”], [0, 1, “-”, 1],
[0, “-”, 0, 0], [0, “-”, 0, 1], [0, “-”, 1, 1], [1, 0, 1, “-”], [1, 0, “-”, 0],
[1, “-”, 0, 0], [“-”, 0, 0, 0], [“-”, 0, 1, 1], [“-”, 1, 0, 0]]
(12.94)

```

This is close to the procedure we want, but we need to think ahead a bit. Recall from the description of the Quine–McCluskey process in the text that, in order to proceed with the second half of the method, we need to know which of the bit strings are prime implicants. That is, which bit strings are never used in a simplification.

We will track which bit strings are used as follows. Before the first loop, we create a set consisting of all of the bit strings in groups. We do this using the functional ‘**union**’ applied to the **entries** of the **groups** table. Note that use of the **nolist** option; by default, **entries** wraps each entry in a list.

```

> }union}(entries(groups, nolist))
{[0, 0, 0, 0], [0, 0, 0, 1], [0, 0, 1, 1], [0, 1, 0, 0], [0, 1, 0, 1], [0, 1, 1, 1],
[1, 0, 0, 0], [1, 0, 1, 0], [1, 0, 1, 1], [1, 1, 0, 0]}
(12.95)

```

Then, each time **MergeBitstrings** returns true, we remove the pair of bit strings from this set, using the **minus** set operator.

The procedure will return the sequence consisting of the next level of bit strings and the prime implicants from this stage. Here is our second attempt at the procedure.

```

1 NextBitListtry2 := proc(lastgroups)
2   local nextL, primeImps, n, A, B, a, b, m;
3   nextL := {};
4   primeImps := ‘union’(entries(lastgroups, nolist));
5   for n from 1 to max(indices(lastgroups)) do
6     A := lastgroups[n];
7     B := lastgroups[n-1];
8     for a in A do
9       for b in B do
10        m := MergeBitstrings(a, b);
11        if m <> false then
12          nextL := nextL union {m};

```

```

13         primeImps := primeImps minus {a,b};
14     end if;
15 end do;
16 end do;
17 end do;
18 nextL := [op(nextL)];
19 return nextL, primeImps;
20 end proc:

```

This still is not sufficient, however, because we also need to update the coverage dictionary as we create new bit strings. Recall that “coverage dictionary” is the name we gave to the table that records, for each bit string, which of the original minterms are covered by that bit string. The coverage dictionary was initialized with the bit strings formed from the minterms.

> *eval(coverageDict)*

```

table ([[0, 1, 0, 0] = {7}, [0, 0, 1, 1] = {8}, [0, 1, 0, 1] = {6},
       [0, 1, 1, 1] = {5}, [1, 1, 0, 0] = {1}, [0, 0, 0, 1] = {9},
       [1, 0, 1, 0] = {3}, [1, 0, 0, 0] = {4}, [0, 0, 0, 0] = {10},
       [1, 0, 1, 1] = {2}])

```

(12.96)

Within the **NextBitList** procedure, we need to update the coverage dictionary. We will make the dictionary a parameter. Note that it will not be necessary to return the updated dictionary, nor is it necessary to make a copy of the parameter. This is because, as we mentioned in Section 2.6 of this manual, tables are “reference types,” so unlike most arguments, they are altered by the procedure.

We update the dictionary within the **m <> false** if statement. When we form a new bit string **m**, we obtain the set of minterms it covers by taking the union of the sets of minterms covered by the two bit strings that were merged. That is,

coverDict[m] := coverDict[a] union coverDict[b];

Note that bit strings formed beyond the first step are typically generated multiple times. However, each time they are generated, they always cover the same set of original minterms.

Here is the final version of **NextBitList**.

```

1 NextBitList := proc (lastgroups, coverDict)
2     local nextL, primeImps, n, A, B, a, b, m;
3     nextL := {};
4     primeImps := 'union'(entries(lastgroups, nolist));
5     for n from 1 to max(indices(lastgroups)) do
6         A := lastgroups[n];
7         B := lastgroups[n-1];
8         for a in A do
9             for b in B do
10                m := MergeBitstrings(a, b);
11                if m <> false then
12                    nextL := nextL union {m};
13                    primeImps := primeImps minus {a, b};

```

```

14         coverDict [m] := coverDict [a] union coverDict [b] ;
15     end if ;
16 end do ;
17 end do ;
18 end do ;
19 nextL := [op (nextL) ] ;
20 return nextL, primeImps ;
21 end proc:

```

We apply it to **groups** to obtain **Fbits1** and **primes1**.

```

> Fbits1, primes1 := NextBitList (groups, coverageDict)
Fbits1, primes1 := [[0,0,0,"-"],[0,0,"-",1],[0,1,0,"-"],
[0,1,"-",1],[0,"-",0,0],[0,"-",0,1],[0,"-",1,1],[1,0,1,"-"],
[1,0,"-",0],[1,"-",0,0],[ "-",0,0,0],[ "-",0,1,1],
[ "-",1,0,0]], ∅

```

(12.97)

We see that there are

```

> nops (Fbits1)
13

```

(12.98)

bit strings in the second level, but no prime implicants coming from the first pass.

```

> nops (primes1)
0

```

(12.99)

In addition, almost as a side effect, the procedure has updated **coverageDict**.

```

> eval (coverageDict)
table ([ [0,1,0,0] = {7} , [0,"-",0,1] = {6,9} , [0,0,1,1] = {8} ,
[0,1,0,1] = {6} , [0,"-",0,0] = {7,10} , [1,0,"-",0] = {3,4} ,
[ "-",0,0,0] = {4,10} , [ "-",0,1,1] = {2,8} , [0,0,"-",1] = {8,9} ,
[0,1,1,1] = {5} , [1,1,0,0] = {1} , [0,0,0,1] = {9} ,
[0,0,0,"-"] = {9,10} , [0,"-",1,1] = {5,8} , [1,"-",0,0] = {1,4} ,
[1,0,1,0] = {3} , [1,0,1,"-"] = {2,3} , [ "-",1,0,0] = {1,7} ,
[0,1,0,"-"] = {6,7} , [1,0,0,0] = {4} , [0,0,0,0] = {10} ,
[0,1,"-",1] = {5,6} , [1,0,1,1] = {2} ])

```

(12.100)

Repeating

Step 4 is to repeat steps 2 and 3.

The **Fbits1** list takes the place of **Fbits**. We apply **sortGroups** to produce **groups1**.

```

> groups1 := sortGroups (Fbits1)
groups1 := table ([0 = {[0,0,0,"-"],[0,"-",0,0],[ "-",0,0,0]} ,
1 = {[0,0,"-",1],[0,1,0,"-"],[0,"-",0,1],[1,0,"-",0],
[1,"-",0,0],[ "-",1,0,0]} , 2 = {[0,1,"-",1],[0,"-",1,1],
[1,0,1,"-"],[ "-",0,1,1]} , 3 = {} , 4 = {}])

```

(12.101)

Then applying **NextBitList** to **groups1** produces **Fbits2** and **primes2**.

```
> Fbits2, primes2 := NextBitList(groups1, coverageDict)
Fbits2, primes2 := [[0, "-", 0, "-"], [0, "-", "-", 1], ["-", "-", 0, 0]],
[1, 0, 1, "-"], [1, 0, "-", 0], ["-", 0, 1, 1]]
```

 (12.102)

We see that we have found three prime implicants. The coverage dictionary was further expanded to include the new bit strings.

```
> eval(coverageDict)
table([[0, 1, 0, 0] = {7}, [0, "-", 0, 1] = {6, 9},
["-", "-", 0, 0] = {1, 4, 7, 10}, [0, 0, 1, 1] = {8}, [0, 1, 0, 1] = {6},
[0, "-", 0, 0] = {7, 10}, [1, 0, "-", 0] = {3, 4}, ["-", 0, 0, 0] = {4, 10},
["-", 0, 1, 1] = {2, 8}, [0, 0, "-", 1] = {8, 9}, [0, 1, 1, 1] = {5},
[1, 1, 0, 0] = {1}, [0, 0, 0, 1] = {9}, [0, 0, 0, "-"] = {9, 10},
[0, "-", "-", 1] = {5, 6, 8, 9}, [0, "-", 1, 1] = {5, 8},
[1, "-", 0, 0] = {1, 4}, [1, 0, 1, 0] = {3}, [1, 0, 1, "-"] = {2, 3},
["-", 1, 0, 0] = {1, 7}, [0, "-", 0, "-"] = {6, 7, 9, 10},
[0, 1, 0, "-"] = {6, 7}, [1, 0, 0, 0] = {4}, [0, 0, 0, 0] = {10},
[0, 1, "-", 1] = {5, 6}, [1, 0, 1, 1] = {2}])
```

 (12.103)

Do the same thing again with **Fbits2**.

```
> groups2 := sortGroups(Fbits2)
groups2 := table([0 = {[0, "-", 0, "-"], ["-", "-", 0, 0]},
1 = {[0, "-", "-", 1]}, 2 = {}, 3 = {}, 4 = {}])
```

 (12.104)

```
> Fbits3, primes3 := NextBitList(groups2, coverageDict)
Fbits3, primes3 := [],
[0, "-", 0, "-"], [0, "-", "-", 1], ["-", "-", 0, 0]]
```

 (12.105)

This time, **Fbits3** was empty, which indicates that no more merging is possible and all prime implicants have been found.

This part of the process concludes by forming the list of all the prime implicants.

```
> allprimeImps := [op(primes1 union primes2 union primes3)]
allprimeImps := [[0, "-", 0, "-"], [0, "-", "-", 1], [1, 0, 1, "-"],
[1, 0, "-", 0], ["-", 0, 1, 1], ["-", "-", 0, 0]]
```

 (12.106)

Forming the Coverage Table

Now that we have identified all of the prime implicants, we will use the coverage dictionary to create the coverage table.

Take a look at the final state of the coverage dictionary.

```
> eval(coverageDict)
```

$$\begin{aligned}
& \text{table}([[0, 1, 0, 0] = \{7\}, [0, "-", 0, 1] = \{6, 9\}, \\
& \quad ["-", "-", 0, 0] = \{1, 4, 7, 10\}, [0, 0, 1, 1] = \{8\}, [0, 1, 0, 1] = \{6\}, \\
& \quad [0, "-", 0, 0] = \{7, 10\}, [1, 0, "-", 0] = \{3, 4\}, ["-", 0, 0, 0] = \{4, 10\}, \\
& \quad ["-", 0, 1, 1] = \{2, 8\}, [0, 0, "-", 1] = \{8, 9\}, [0, 1, 1, 1] = \{5\}, \\
& \quad [1, 1, 0, 0] = \{1\}, [0, 0, 0, 1] = \{9\}, [0, 0, 0, "-"] = \{9, 10\}, \\
& \quad [0, "-", "-", 1] = \{5, 6, 8, 9\}, [0, "-", 1, 1] = \{5, 8\}, \\
& \quad [1, "-", 0, 0] = \{1, 4\}, [1, 0, 1, 0] = \{3\}, [1, 0, 1, "-"] = \{2, 3\}, \\
& \quad ["-", 1, 0, 0] = \{1, 7\}, [0, "-", 0, "-"] = \{6, 7, 9, 10\}, \\
& \quad [0, 1, 0, "-"] = \{6, 7\}, [1, 0, 0, 0] = \{4\}, [0, 0, 0, 0] = \{10\}, \\
& \quad [0, 1, "-", 1] = \{5, 6\}, [1, 0, 1, 1] = \{2\}])
\end{aligned} \tag{12.107}$$

Each bit string, and in particular each prime implicant, is an index in this table. The corresponding entry is the set of integers which are the indices to the original minterms in **Fbits**. Thus, to determine which of the original minterms are covered by each prime implicant, we look it up in the table.

We will model the coverage table as a matrix. Each row corresponds to a prime implicant, so there will be

$$\begin{aligned}
& > \text{nops}(\text{allprimeImps}) \\
& \quad 6
\end{aligned} \tag{12.108}$$

rows. Each column corresponds to a minterm, so there are

$$\begin{aligned}
& > \text{nops}(\text{Fbits}) \\
& \quad 10
\end{aligned} \tag{12.109}$$

columns. The entries in the matrix will be 0s and 1s with 1 in position (i, j) indicating that the prime implicant at position i in **allprimeImps** covers the minterm at position j in **Fbits**.

Recall that if the **Matrix** command is given two integers as its only arguments, it will create the matrix whose size is specified by the integers and has all 0 entries.

$$\begin{aligned}
& > \text{Matrix}(\text{nops}(\text{allprimeImps}), \text{nops}(\text{Fbits})) \\
& \quad \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}
\end{aligned} \tag{12.110}$$

To enter 1s in the appropriate positions, we loop over the rows, considering each prime implicant in turn. For each prime implicant, we look up its entry in the coverage dictionary to obtain the set of minterms it covers. For each of those minterms, we place a 1 in the matrix.

The following procedure initializes the coverage table.

```

1  initCoverMatrix := proc (minterms, primeImps, coverDict)
2      local M, i, C, j;
3      M := Matrix (nops (primeImps), nops (minterms));
4      for i from 1 to nops (primeImps) do
5          C := coverDict [primeImps [i]];
6          for j in C do
7              M[i, j] := 1;
8          end do;
9      end do;
10     return M;
11 end proc;

```

Applied to our example, this produces the following coverage table.

> *coverageTable* := *initCoverMatrix* (*Fbits*, *allprimeImps*, *coverageDict*)

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 1 & 1 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 1 & 1 & 0 & 1 & 1 & 0 \\ 0 & 1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 1 & 0 & 0 & 1 & 0 & 0 & 1 \end{bmatrix}$$

(12.111)

Manipulating the Matrix

Once the coverage table is set up, we move to steps 6 and 7, determining which prime implicants to include in the minimal expression. In step 6, we identify the essential prime implicants and in step 7 we identify which of the nonessential prime implicants we will include. We will see how to identify the prime implicants to use in a moment.

To aid in performing both steps 6 and 7, we will be manipulating the coverage table. Once we have decided to include a particular prime implicant in the minimal expression, we can take three actions.

First, record the decision by adding the prime implicant to a new list, say **minBits**, the list of bit strings to be included in the minimal expression.

Second, delete that prime implicant's row from the coverage table and delete the columns corresponding to the minterms it covered. We know the prime implicant will be in the expression, and, thus the minterms it covers are satisfied. Hence, there is no longer any need to keep track of that information.

Third, delete the prime implicant and the minterms it covers from the lists storing them (**Fbits** and **allprimeImps**). This is to ensure that the indices of **Fbits** and **allprimeImps** continue to match the row and column numbers of the matrix.

We write a procedure that implements these actions. Its input will be the index to the prime implicant that has been chosen. It will also accept the names of the coverage matrix, the list of minterms, and the list of prime implicants. All of these will be modified in the procedure (refer to

Modifying Arguments above). The procedure will return the bit string of the prime implicant that was chosen.

Our procedure will be called **UpdateCT**, for “update coverage table.” The **minBits** list, the list of chosen prime implicants, will be updated via the return value. This accomplishes the first task for this procedure.

Second, we must delete the row corresponding to the chosen prime implicant and the columns corresponding to the minterms covered by that implicant. Suppose, in our example, that we have decided to include the fourth prime implicant in the final result. This is

```
> allprimeImps[4]
    [1, 0, “–”, 0]
```

(12.112)

From **coverageTable**, we need to remove row 4 (since this corresponds to the prime implicant). We also need to remove the columns corresponding to the minterms covered by this prime implicant. To determine which columns are to be removed, we find the locations of the 1s in the row of the matrix.

To determine the locations of the 1s, we loop over the columns checking each position in row 4 to see if it is 1 or not. We use the **ColumnDimension** command from the **LinearAlgebra** package to determine the number of columns.

```
> with(LinearAlgebra):

> covered := {}
    covered := ∅

> for i to ColumnDimension(coverageTable) do
    if coverageTable[4, i] = 1 then
        covered := covered union {i}
    end if
end do :
covered
    {3, 4}
```

(12.114)

We now know that we need to remove row 4 and columns 3 and 4. To do this, we use a complicated selection.

We have seen that ranges can be used to select from lists. The same is true for matrices. For example, we can obtain the first three rows of this matrix as follows.

```
> coverageTable[1..3, 1..–1]
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 1 & 1 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 1 & 1 & 0 & 1 & 1 & 0 \\ 0 & 1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

(12.115)

The first range indicates rows 1 through 3, the second that we want all the columns, from the first to the last.

You can also give lists of the rows instead of ranges.

$$\begin{aligned}
 &> \text{coverageTable}[[1, 2, 3], [1, 2, 3, 4, 5, 6, 7, 8, 9, 10]] \\
 &\quad \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 1 & 1 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 1 & 1 & 0 & 1 & 1 & 0 \\ 0 & 1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}
 \end{aligned} \tag{12.116}$$

In our example, we want all of the rows except for the fourth, which we can obtain from the pair of ranges **1..3** and **5..6**. More generally, if **newPI** is the index of the new prime implicant to be included in the minimal expression, we would use **1..(newPI-1)** and **(newPI+1)..-1**.

$$\begin{aligned}
 &> \text{row4} := 4 \\
 &\quad \text{row4} := 4
 \end{aligned} \tag{12.117}$$

$$\begin{aligned}
 &> \text{coverageTable}[[1 ..(\text{row4} - 1), (\text{row4} + 1) .. - 1], [1 .. - 1]] \\
 &\quad \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 1 & 1 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 1 & 1 & 0 & 1 & 1 & 0 \\ 0 & 1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 1 & 0 & 0 & 1 & 0 & 0 & 1 \end{bmatrix}
 \end{aligned} \tag{12.118}$$

For the columns we will take a different approach. Begin with the set of all column indexes.

$$\begin{aligned}
 &> \text{colList} := \{\$1 .. \text{ColumnDimension}(\text{coverageTable})\} \\
 &\quad \text{colList} := \{1, 2, 3, 4, 5, 6, 7, 8, 9, 10\}
 \end{aligned} \tag{12.119}$$

Now remove from this set the columns that are to be removed by subtracting the **covered** set.

$$\begin{aligned}
 &> \text{colList} := \text{colList} \text{ minus covered} \\
 &\quad \text{colList} := \{1, 2, 5, 6, 7, 8, 9, 10\}
 \end{aligned} \tag{12.120}$$

Then turn it into a list.

$$\begin{aligned}
 &> \text{colList} := [\text{op}(\text{colList})] \\
 &\quad \text{colList} := [1, 2, 5, 6, 7, 8, 9, 10]
 \end{aligned} \tag{12.121}$$

Using this to select the columns, we obtain the desired matrix.

$$\begin{aligned}
 &> \text{coverageTable}[[1 ..(\text{row4} - 1), (\text{row4} + 1) .. - 1], \text{colList}] \\
 &\quad \begin{bmatrix} 0 & 0 & 0 & 1 & 1 & 0 & 1 & 1 \\ 0 & 0 & 1 & 1 & 0 & 1 & 1 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 & 1 & 0 & 0 & 1 \end{bmatrix}
 \end{aligned} \tag{12.122}$$

The reader is encouraged to compare this to the original matrix. Note that in this example, we have not actually modified the table.

The last tasks are to remove the selected prime implicant from the list of prime implicants, and remove the covered minterms from **Fbits**. We use **subsop** to remove elements from lists, as usual. For instance, the following removes the fourth prime implicant.

```
> subsop(4 = NULL, allprimeImps)
[[0, "-", 0, "-"], [0, "-", "-", 1], [1, 0, 1, "-"], ["-", 0, 1, 1],
["-", "-", 0, 0]]
```

(12.123)

When removing the minterms, we remove them in the reverse order. For instance, to remove the minterms in locations 3 and 4, we first remove the minterm in position 4 and then the minterm in position 3. Otherwise, if we remove the minterm in location 3 first, then all the other minterms shift location by 1. That is, the minterm previously in location 4 is now in location 3. By removing them in reverse order, this is not a concern.

Here is the procedure. Remember that since we are using the **evaln** parameter option in order to make parameters modifiable, we must use **eval** when we need the values of those objects.

```
1 UpdateCT := proc (newPI, coverTable::evaln, minterms::evaln,
2   primeImps::evaln)
3   local newPIbits, numcols, covered, i, colList;
4   newPIbits := eval (primeImps) [newPI];
5   numcols := LinearAlgebra[ColumnDimension] (eval (coverTable));
6   covered := {};
7   for i from 1 to numcols do
8     if eval (coverTable) [newPI, i] = 1 then
9       covered := covered union {i};
10    end if;
11  end do;
12  colList := [op({$1..numcols} minus covered)];
13  coverTable :=
14    eval (coverTable) [[1..(newPI-1), (newPI+1)..-1], colList];
15  primeImps := subsop (newPI=NULL, eval (primeImps));
16  for i from nops (covered) to 1 by -1 do
17    minterms := subsop (covered[i]=NULL, eval (minterms));
18  end do;
19  return newPIbits;
20 end proc;
```

Finding Essential Prime Implicants

Next we write a procedure to identify the essential prime implicants. Recall that a prime implicant is essential when it is the only prime implicant to cover some minterm. In terms of the coverage table, this is equivalent to the existence of a column with only one 1.

We will locate the essential prime implicants as follows. First, we initialize the set of essential prime implicants to the empty list.

We proceed in a manner similar to the **MergeBitstrings** procedure. We use a for loop to step through the columns of the coverage table. Within this loop, we initialize a name, **rowhas1**, to 0.

We then enter a second for loop to step through the entries in the columns. When a 1 entry has been found, we check **rowhas1**. If that name is 0, then it is assigned to the current row number. If it is not 0, then we have found a second 1 in the column and we assign **rowhas1** to -1 and use **break** to terminate the inner loop. After the inner loop, we test **rowhas1**. If it is positive, then we know that only one 1 was located in that column, and hence the row the solitary 1 was found in corresponds to an essential prime implicant. In this case, we add the row number (**rowhas1**) to **essentials**.

The following procedure implements this algorithm and returns the list of essential prime implicants.

```

1 FindEssentials := proc (coverTable)
2   local essentials, i, j, rowhas1;
3   essentials := {};
4   for i from 1 to LinearAlgebra[ColumnDimension] (coverTable) do
5     rowhas1 := 0;
6     for j from 1 to LinearAlgebra[RowDimension] (coverTable) do
7       if coverTable[j, i] = 1 then
8         if rowhas1 = 0 then
9           rowhas1 := j;
10        else
11          rowhas1 := -1;
12          break;
13        end if;
14      end if;
15    end do;
16    if rowhas1 > 0 then
17      essentials := essentials union {rowhas1};
18    end if;
19  end do;
20  return essentials;
21 end proc;
```

We use this to determine the essential prime implicants of our example.

```

> essentialPIs := FindEssentials(coverageTable)
essentialPIs := {2,6} (12.124)
```

Now that we have the essential prime implicants, we can initialize **minBits** and apply **UpdateCT** to the essential prime implicants. Once again, we loop through the list backwards.

```

> minBits := []
minBits := [] (12.125)

> for i from nops(essentialPIs) by -1 to 1 do
minBits := [op(minBits), UpdateCT(essentialPIs[i], coverageTable,
Fbits, allprimeImps)]
end do;
```

> *minBits*
 $[[\text{"-"}, \text{"-"}, 0, 0], [0, \text{"-"}, \text{"-"}, 1]]$ (12.126)

> *coverageTable*

$$\begin{bmatrix} 0 & 0 \\ 1 & 1 \\ 0 & 1 \\ 1 & 0 \end{bmatrix}$$
 (12.127)

Completing the Coverage

Provided that the essential prime implicants did not completely cover the original minterms, we must complete the coverage with nonessential prime implicants. First, we ensure that the coverage is not complete by checking the column dimension.

> *evalb (LinearAlgebra[ColumnDimension] (coverageTable) > 0)*
true (12.128)

As we mentioned earlier, we will use a heuristic approach to find a minimal set of prime implicants rather than using an exhaustive search to determine the minimum. The heuristic we use will be to choose the prime implicant with the most extensive coverage of the remaining minterms.

To find such a prime implicant, we will do the following. First, initialize **maxCoverage** and **bestImp** both to 0. Then loop over each row of the (modified) coverage table. For each row, we will compute the sum of the entries. If this sum is greater than **maxCoverage**, then set **maxCoverage** to the sum and set **bestImp** to the row number. Once the loop is complete, **bestImp** will be the index to a row with maximum coverage and will be the next prime implicant added to the **minBits** list.

Here is the procedure that implements this strategy.

```

1  findBestImp := proc (coverTable)
2      local maxCoverage, bestImp, i, j, sum;
3      maxCoverage := 0;
4      bestImp := 0;
5      for i from 1 to LinearAlgebra[RowDimension] (coverTable) do
6          sum := 0;
7          for j from 1 to LinearAlgebra[ColumnDimension] (coverTable) do
8              sum := sum + coverTable[i, j];
9          end do;
10         if sum > maxCoverage then
11             maxCoverage := sum;
12             bestImp := i;
13         end if;
14     end do;
15     return bestImp;
16 end proc;
```

As long as the coverage table is not empty, we apply this procedure to it to obtain the next implicant. We add the implicant to the list **minBits** representing the minimal expression and update the coverage table using **updateCT**.

```
> while LinearAlgebra[ColumnDimension](coverageTable) > 0 do
    nextPI := findBestImp(coverageTable);
    minBits := [op(minBits), UpdateCT(nextPI, coverageTable, Fbits,
        allprimeImps)]
end do
nextPI := 2
minBits := [[ "-", "-", 0, 0], [0, "-", "-", 1], [1, 0, 1, "-"]]
```

(12.129)

All that is left is to translate **minBits** back into a logical expression. This can be done using **BitStringtoMT** applied to each element of **minBits** with the **map** command and then combined into one expression with **&or**.

```
> &or (op (map (BitStringtoMT, minBits, [w, x, y, z])))
((¬y) ∧ (¬z)) ∨ ((¬w) ∧ z) ∨ (w ∧ (¬x) ∧ y)
```

(12.130)

Putting It All Together

Finally, we assemble the pieces into a single procedure, which accepts a logical expression in disjunctive normal form and a list of its variables. It returns a minimal equivalent expression.

```
1 QuineMcCluskey := proc (F, variables)
2     local Fbits, FbitsL, coverageDict, groups, primes, i,
3         allprimeImps, j, coverageTable, essentialPIs, minBits,
4         nextPI;
5     uses Logic;
6     Fbits := DNFTtoBitList (F, variables);
7     coverageDict := initCoverDict (Fbits);
8     i := 0;
9     FbitsL[0] := Fbits;
10    while FbitsL[i] <> [] do
11        i := i + 1;
12        groups[i] := sortGroups (FbitsL[i-1]);
13        FbitsL[i], primes[i] :=
14            NextBitList (groups[i], coverageDict);
15    end do;
16    allprimeImps := {};
17    for j from 1 to i do
18        allprimeImps := allprimeImps union primes[j];
19    end do;
20    allprimeImps := [op(allprimeImps)];
21    coverageTable :=
22        initCoverMatrix (Fbits, allprimeImps, coverageDict);
23    essentialPIs := FindEssentials (coverageTable);
24    minBits := [];
25    for i from nops (essentialPIs) to 1 by -1 do
```

```

22     minBits := [op(minBits), UpdateCT(essentialPIs[i],
23         coverageTable, Fbits, allprimeImps)];
24 end do;
25 while LinearAlgebra[ColumnDimension](coverageTable) > 0 do
26     nextPI := findBestImp(coverageTable);
27     minBits := [op(minBits),
28         UpdateCT(nextPI, coverageTable, Fbits, allprimeImps)];
29 end do;
30 return &or(op(map(BitStringtoMT, minBits, variables)));
31 end proc:

```

Define **Ex10** to be the expression in Example 10 from Section 12.4 of the text.

$$\begin{aligned}
 &> E10 := (w \&\text{and } x \&\text{and } y \&\text{and } \&\text{not}(z)) \&\text{or} \\
 &\quad (w \&\text{and } \&\text{not}(x) \&\text{and } y \&\text{and } z) \&\text{or} \\
 &\quad (w \&\text{and } \&\text{not}(x) \&\text{and } y \&\text{and } \&\text{not}(z)) \&\text{or} \\
 &\quad (\&\text{not } w \&\text{and } x \&\text{and } y \&\text{and } z) \&\text{or} \\
 &\quad (\&\text{not } w \&\text{and } x \&\text{and } \&\text{not}(y) \&\text{and } z) \&\text{or} \\
 &\quad (\&\text{not } w \&\text{and } \&\text{not}(x) \&\text{and } y \&\text{and } z) \&\text{or} \\
 &\quad (\&\text{not } w \&\text{and } \&\text{not}(x) \&\text{and } \&\text{not}(y) \&\text{and } z) \\
 &E10 := (w \wedge x \wedge y \wedge (\neg z)) \vee (w \wedge (\neg x) \wedge y \wedge z) \vee \\
 &\quad (w \wedge (\neg x) \wedge y \wedge (\neg z)) \vee (\neg w \wedge x \wedge y \wedge z) \vee \\
 &\quad (\neg w \wedge x \wedge (\neg y) \wedge z) \vee (\neg w \wedge (\neg x) \wedge y \wedge z) \vee \\
 &\quad (\neg w \wedge (\neg x) \wedge (\neg y) \wedge z)
 \end{aligned} \tag{12.131}$$

$$\begin{aligned}
 &> QuineMcCluskey(E10, [w, x, y, z]) \\
 &\quad (w \wedge y \wedge (\neg z)) \vee ((\neg w) \wedge z) \vee (w \wedge (\neg x) \wedge y)
 \end{aligned} \tag{12.132}$$

Note that this is the first of the two answers given in the solution to Example 10.

Solutions to Computer Projects and Computations and Explorations

Computer Projects 2

Construct a table listing the set of values of all 256 Boolean functions of degree three.

Solution: The Boolean functions of degree three are in one-to-one correspondence with the subsets of $\{true, false\}^3$. This is because each subset S of $\{true, false\}^3$ can be identified with the unique Boolean function of degree three which returns true on the members of S and false on all other inputs.

Thus, we begin by constructing the set $\{true, false\}^3$ and its power set.

To construct $\{true, false\}^3$, we will use the **cartprod** command from the **combinat** package. (Refer to Section 2.1 of this manual for details.) Like many of the commands for generating combinatorial objects, **cartprod** produces a list with indices **finished** and **nextvalue**. We use it to form the set **TF3**.

$$\begin{aligned}
 &> TF3 := \{ \\
 &\quad TF3 := \emptyset
 \end{aligned} \tag{12.133}$$

```

> TF3iterator := combinat[cartprod]([true,false]$3):
> while not TF3iterator[finished] do
  TF3 := TF3 union{TF3iterator[nextvalue]() }
end do :
> TF3
{[false,false,false],[false,false,true],[false,true,false],
 [false,true,true],[true,false,false],[true,false,true],
 [true,true,false],[true,true,true]}

```

(12.134)

To produce the subsets of $\{true,false\}^3$, we use the **subsets** command from **combinat**. This command also produces a table with indices **finished** and **nextvalue**.

```

> TF3subsets := combinat[subsets](TF3):

```

Now we will create a list of all of the Boolean functions. The subsets that are produced by **TF3subsets** are each valid inputs to the **BooleanFromTable** procedure we wrote in Section 12.2. We also apply **BooleanSimplify**, in order to have simpler representations, before adding the expression to the list.

```

> allFunctions := []
allFunctions := []
> while not TF3subsets[finished] do
  nextTF3subset := TF3subsets[nextvalue]();
  nextTF3function := BooleanFromTable(nextTF3subset,[x,y,z]);
  nextTF3function := BooleanSimplify(nextTF3function);
  allFunctions := [op(allFunctions),nextTF3function];
end do :

```

(12.135)

The **allFunctions** list is lengthy, so we display only a few members.

```

> nops(allFunctions)
256

```

(12.136)

```

> allFunctions[1..10]
[false, (¬x) ∧ (¬y) ∧ (¬z), z ∧ (¬x) ∧ (¬y), y ∧ (¬x) ∧ (¬z), y ∧ z ∧ (¬x),
 x ∧ (¬y) ∧ (¬z), x ∧ z ∧ (¬y), x ∧ y ∧ (¬z), x ∧ y ∧ z,
 (¬x) ∧ (¬y)]

```

(12.137)

To obtain the values of the functions, we will use the **TruthTable** command. Recall from the first section that **TruthTable** requires two arguments, a Boolean expression and a list of variables that appear in the expression. The result is a **DataFrame** object. Using the **output** option, we can specify the output to be a **table** with indices lists of truth values and corresponding entries the value of the expression at the index.

We form the list of the truth tables associated to each function in the **allFunctions** list with the **map** command. Note that we use the third argument **[x,y,z]** and fourth argument **output=table** in **map** so that this list of variables and the option are passed to **TruthTable** each time it is applied to an expression from **allFunctions**.

```
> allTables := map(TruthTable, allFunctions, [x, y, z], output = table) :
```

Each entry in **allTables** is a table storing the truth table.

```
> allTables[136]
table([(true,false,false) = false, (false,false,true) = true,
      (true,false,true) = true, (true,true,false) = true,
      (true,true,true) = false, (false,true,true) = false,
      (false,true,false) = true, (false,false,false) = false]) (12.138)
```

The output below indicates that the value of the 136th function on *(true, false, true)* is *true*.

```
> allTables[136][true,false,true]
true (12.139)
```

To display the entire truth table, loop through the members of **TF3**, each of which corresponds to a row of the truth table. Then, loop through members of **allTables** and obtain the value of the corresponding function. We must apply **op** to the member of **TF3**, since the index to a truth table is a sequence not a list. For demonstration purposes, we restrict to the functions 100 through 105.

```
> for i from 1 to nops(TF3) do
  row := TF3[i];
  for j from 100 to 105 do
    row := row, allTables[j][op(TF3[i])]
  end do;
  print(row);
end do :
[false,false,false], true, true, true, true, true, true
[false,false,true], true, true, true, true, true, true
[false,true,false], false, false, false, false, false, false
[false,true,true], true, true, true, false, false, false
[true,false,false], false, false, false, true, true, true
[true,false,true], true, false, false, true, false, false
[true,true,false], false, true, false, false, true, false
[true,true,true], false, false, true, false, false, true (12.140)
```

The output above indicates that on the input *(false, false, false)*, all six Boolean functions associated to the integers 100 through 105 output *true*.

Computations and Explorations 6

Randomly generate 10 different Boolean expressions in four variables and determine the average number of steps required to minimize them using the Quine–McCluskey method.

Solution: To solve this problem, we need to find a way to generate random Boolean expressions and, then, we must find a method of examining the minimization process so that we can count the number of steps.

In the **Logic** package, the command **Random** produces a random Boolean expression. The only required argument is a set or list specifying the symbols to be used. For example, to produce a random Boolean expression on the symbols u , v , w , x , y , and z , you enter the following.

```
> Random([u, v, w, x, y, z])
(u ∧ v ∧ w ∧ y ∧ (¬x) ∧ (¬z)) ∨ (u ∧ w ∧ x ∧ y ∧ z ∧ (¬v)) ∨
(u ∧ w ∧ (¬v) ∧ (¬x) ∧ (¬y) ∧ (¬z)) ∨
(v ∧ w ∧ (¬u) ∧ (¬x) ∧ (¬y) ∧ (¬z)) ∨
(v ∧ y ∧ (¬u) ∧ (¬w) ∧ (¬x) ∧ (¬z)) ∨
(v ∧ (¬u) ∧ (¬w) ∧ (¬x) ∧ (¬y) ∧ (¬z))
```

(12.141)

The **Random** command also accepts a second optional argument: **form=CNF**, **form=DNF**, or **form=MOD2**, specifying the form of the expression produced. The default form is disjunctive normal form.

Having determined how to generate random expressions, we need to find a way to count the number of steps taken during the minimization process. There are (at least) three approaches we could take to this part of the problem.

The first is to measure the time taken to execute the procedure. We have done this many times before.

```
> QMtime := []:
for i to 10 do
    randExp := Random([u, v, w, x, y, z]);
    st := time();
    QuineMcCluskey(randExp, [u, v, w, x, y, z]);
    et := time() - st;
    QMtime := [op(QMtime), et]
end do:
Statistics[Mean](QMtime)
0.0008000000000000000
```

(12.142)

The second approach is to modify the procedure to count the number of times certain operations are called. For example, we may be interested in the number of times that the **UpdateCT** procedure is executed. In this case, we can alter **UpdateCT** to include a global variable that is incremented at the start of every execution.

```
1 UpdateCT := proc (newPI, coverTable::evaln, minterms::evaln,
2   primeImps::evaln)
3   local newPIbits, numcols, covered, i, colList;
4   global countUpdateCT;
5   countUpdateCT := countUpdateCT + 1;
6   newPIbits := eval (primeImps) [newPI];
7   numcols := LinearAlgebra[ColumnDimension] (eval (coverTable));
8   covered := {};
9   for i from 1 to numcols do
10    if eval (coverTable) [newPI, i] = 1 then
        covered := covered union {i};
    end if;
  end for;
end proc;
```

```

11     end if ;
12 end do;
13 collist := [op({$1..numcols} minus covered)];
14 coverTable :=
    eval(coverTable) [[1..(newPI-1), (newPI+1)..-1], collist];
15 primeImps := subsop(newPI=NULL, eval(primeImps));
16 for i from nops(covered) to 1 by -1 do
17     minterms := subsop(covered[i]=NULL, eval(minterms));
18 end do;
19 return newPIbits;
20 end proc:

```

We must initialize the variable to 0.

```

> countUpdateCT := 0
    countUpdateCT := 0

```

(12.143)

Now execute **QuineMcCluskey** on 100 random expressions.

```

> for i to 100 do
    randExp := Random([u, v, w, x, y, z]);
    QuineMcCluskey(randExp, [u, v, w, x, y, z])
end do :

```

The **countUpdateCT** variable will now store the number of times **UpdateCT** was called. Dividing by 10 gives us the average.

```

>  $\frac{\text{countUpdateCT}}{100.}$ 
    4.2900000000

```

(12.144)

The third approach is to make use of Maple's debugging facilities. We used **trace** in earlier chapters to get information about the workings of a procedure. Here, we will use the **showstat** command.

If you apply **showstat** to the name of a procedure, it will display the definition of the procedure with line numbers added on the left hand side. An integer or a range of integers can be given as a second argument to narrow the display to the desired lines.

```

> showstat (QuineMcCluskey, 5..8)
QuineMcCluskey := proc (F, variables)
    local Fbits, FbitsL, coverageDict, groups, primes, i, allprimeImps,
    j, coverageTable, essentialPIs, minBits, nextPI;
    ...
    5 while FbitsL[i] <> [] do
    6 i := i+1;
    7 groups[i] := sortGroups(FbitsL[i-1]);
    8 FbitsL[i], primes[i] := NextBitList(groups[i], coverageDict)
    end do;
    ...
end proc

```

You can get more information by telling Maple to track the procedure. You do this by calling the **debugopts** command with argument **traceproc=** and then the name of the procedure.

```
> debugopts (traceproc = QuineMcCluskey)
```

Now we apply **QuineMcCluskey** to 1000 random expressions. We suppress the output as it is not needed here.

```
> for i to 1000 do
    randExp := Random([u,v,w,x,y,z]);
    QuineMcCluskey(randExp, [u,v,w,x,y,z])
end do :
```

If we call the **showstat** command again, the output gives us more information than it did before.

```
> showstat (QuineMcCluskey)
```

```
QuineMcCluskey := proc (F, variables)
    local Fbits, FbitsL, coverageDict, groups, primes, i, allprimeImps,
j, coverageTable, essentialPIs, minBits, nextPI;
    | Calls Seconds Words |
PROC | 1000 0.771 11 315 189 |
1 | 1000 0.131 1 769 082 | Fbits := DNFtoBitList (F, variables);
2 | 1000 0.007 361 562 | coverageDict := initCoverDict (Fbits);
3 | 1000 0.001 0 | i := 0;
4 | 1000 0.002 273 000 | FbitsL[0] := Fbits;
5 | 1000 0.013 14 080 | while FbitsL[i] <> [] do
6 | 1816 0.003 0 | i := i+1;
7 | 1816 0.276 5 184 118 | groups[i] := sortGroups (FbitsL[i-1]);
8 | 1816 0.075 897 695 | FbitsL[i], primes[i] := NextBitList (groups[i], cove
do;
9 | 1000 0.000 0 | allprimeImps := {};
10 | 1000 0.002 0 | for j to i do
11 | 1816 0.010 31 134 | allprimeImps := allprimeImps union primes[j] end
do;
12 | 1000 0.001 4000 | allprimeImps := [op(allprimeImps)];
13 | 1000 0.036 554 955 | coverageTable := initCoverMatrix (Fbits, allprimeI
14 | 1000 0.031 330 035 | essentialPIs := FindEssentials (coverageTable);
15 | 1000 0.003 0 | minBits := [];
16 | 1000 0.014 0 | for i from nops (essentialPIs) by -1 to 1 do
17 | 4187 0.096 961 581 | minBits := [op(minBits), UpdateCT (essentialPIs[i]
do;
18 | 1000 0.004 20 100 | while 0 < LinearAlgebra[LinearAlgebra:-
ColumnDimension] (coverageTable) do
19 | 5 0.000 430 | nextPI := findBestImp (coverageTable);
20 | 5 0.000 895 | minBits := [op(minBits), UpdateCT (nextPI, coverageTable,
do;
21 | 1000 0.066 912 522 | return Logic:-'&or' (op (map (BitStringtoMT, minBit
proc
```

In addition to the line numbers, you see three columns labeled Calls, Seconds, and Words. Note that above line 1 is line PROC. This refers to information for the procedure as a whole.

The Calls column reports the number of times the line was executed. That the Calls column contains 10 in the PROC row indicates that the procedure was called 1000 times. Note that lines 17 and 20 are the lines containing the calls to **UpdateCT**. The sum of the values in the Count column is the number of times **UpdateCT** was called.

The Seconds column reports the amount of CPU time that was spent executing the line.

The Words column indicates the amount of memory that was allocated as a result of the statement.

Together, the three columns give you a considerable amount of information about the computational complexity, performance, and memory requirements of a procedure. In **QuineMcCluskey**, we see that the most time and memory are used in line 7, the call to **sortGroups**. However, the **UpdateCT** procedure was executed more often in line 17.

You can also have Maple store this information in a table for you by calling `debugopts` with argument **traceproctable=** and the name of the procedure.

```
> traceTable := debugopts(traceproctable = QuineMcCluskey)
```

$$\text{traceTable} := \left[\begin{array}{l} 1 \text{..}22 \times 1 \text{..}3 \text{ Array} \\ \text{Data Type: integer}_4 \\ \text{Storage: rectangular} \\ \text{Order: C_order} \end{array} \right] \quad (12.145)$$

The entries in this table correspond to the information displayed. The first row stores information about the procedure as a whole.

```
> traceTable[1, 1..3]
[ 1000 771 11315189 ] (12.146)
```

Information on specific lines is stored in the row one greater than the line number. For instance, the data related to the calls of **UpdateCT**, in line 17 and 20, are contained in the table at 18 and 21.

```
> traceTable[[18, 21], 1..3]
[ 4187 96 961581 ]
[ 5 0 895 ] (12.147)
```

Note that the second column, containing the time measurement, has been multiplied by 1000.

Executing

```
> debugopts(traceproc = QuineMcCluskey)
```

a second time clears the information that was stored and toggles the option to have Maple record the Calls, Seconds, and Words information back off.

Exercises

Exercise 1. Use Maple to verify De Morgan's Laws and the commutative and associative laws. (See Table 5 of Section 12.1.)

Exercise 2. Construct truth tables for each of the following pairs of Boolean expressions and decide whether they are logically equivalent.

- a) $a \rightarrow b$ and $b \rightarrow a$
- b) $a \rightarrow \bar{b}$ and $b \rightarrow \bar{a}$
- c) $a + bc$ and $(a + b + d)(a + c + d)$.

Exercise 3. Write a Maple procedure that translates Boolean expressions or functions written in terms of 0, 1, and the inert addition, multiplication, and negation operators into expressions in terms of the logical operators and values. (Hint: the implementation of the InfixToTree procedure in Chapter 11 of this manual may be helpful.)

Exercise 4. Write a Maple procedure that, given a Boolean function, represents this function using only the **&nand** operator.

Exercise 5. Use the procedure in the previous exercise to represent the following Boolean functions using only the **&nand** operator.

- a) $F(x, y, z) = xy + \bar{y}z$
- b) $G(x, y, z) = x + \bar{x}y + \bar{y}z$
- c) $H(x, y, z) = xyz + \bar{x}y\bar{z}$

Exercise 6. Write a Maple procedure that, given a Boolean function, represents this function using the **&nor** operator.

Exercise 7. Use the procedure in the previous exercise to represent the Boolean functions in Exercise 5 using only the **&nor** operator.

Exercise 8. Write a Maple procedure for determining the output of a threshold gate, given the values of n Boolean variables as input, and given the threshold value and a set of weights for the threshold gate. (See the Supplementary Exercises of Chapter 12 for information on threshold gates.)

Exercise 9. Develop a Maple procedure that, given a Boolean function in four variables, determines whether it is a threshold function, and if so, finds the appropriate threshold gate representing this function. (See the Supplementary Exercises of Chapter 12.)

Exercise 10. A Boolean expression e is called self dual if it is logically equivalent to its dual e^d . Write a Maple procedure to test whether a given expression is self dual.

Exercise 11. Determine, for each integer $n \in \{1, 2, 3, 4, 5, 6\}$, the total number of Boolean functions of n variables and the number of those functions that are self dual.

Exercise 12. Write a Maple procedure that, given a positive integer n , constructs a list of all Boolean functions of degree n . Use your procedure to find all Boolean functions of degree 4.

Exercise 13. Use **DontCare** to compute a minimal sum of products expansion for the Boolean functions with don't care conditions specified by the Karnaugh maps shown in Exercises 30 through 32 of Section 12.4.

Exercise 14. How can you change exactly one character in the definition of the procedure **DontCare** so that it returns the last expression of minimum length that it encounters that is equivalent to the input function? (As written now, it returns the first.)

Exercise 15. Use the procedure you wrote in Exercise 10 to write a Maple procedure to generate random Boolean expressions in 4 variables and stop when it has found one that is self dual. Run the program several times and time it. Find the average time. Repeat for Boolean expressions in 5 and 6 variables. Can you make any conjectures from this information?

Exercise 16. Revise the procedure **DontCare** to return all minimal expressions that it finds, rather than just the first.

Exercise 17. Revise the procedure **DontCare** to use different measures of complexity of Boolean expressions, such as the number of Boolean operations, etc.

Exercise 18. Modify **QuineMcCluskey** to allow for don't care conditions. See the discussion at the end of the Don't Care Conditions subsection in Section 12.4 of this manual.

Exercise 19. Modify **QuineMcCluskey** to use backtracking instead of the heuristic approach in order to determine the expression with the minimum number of terms. Use a large number of randomly generated expressions to compare the old procedure with the new and determine how often the heuristic produces nonoptimal output.